

# 跟我学 BASIC

苹果 I 型  
LASER310型 微机使用

天津市青少年科技辅导员协会  
天津市青少年科技中心编写

学术书刊出版社

## 前 言

一、这本书是为中小学青少年课外开展电子计算机活动，学习使用BASIC语言而编写的。内容简明扼要，通俗易懂，适合于在青少年科技辅导员、中学生、小学生和其他初学者，速成普及使用。

二、由于不同类型的微型计算机所使用的BASIC语言存在一些差别，本教材以在APPLE II PLUS微型机上使用为基础，兼顾LASER-310微型机，对在LASER-310型机上使用时的差别加以说明。参考答案是按APPLE II型机做的。

三、这本书分五章。针对初学者的特点，贯彻理论联系实际的原则，注意上机操作，将基本BASIC语言和扩展BASIC语言横向分为三轮，使初学者能由浅入深逐步展开进行学习。如果做为教材使用，讲课与上机的时间大致是一比一，可在八十课时内学完。

对于小学和初中学生，可根据他们的年龄特征和文化水平选择本教材部分内容学习。如小学生可学习第一章，第二章一至二十二节和二十三节中的随机函数、取整函数、二十八至三十节以及第四章等内容。

四、这本书附有习题及参考答案。每次课后的习题，不仅复习、巩固这次课的内容，而且为下次课做些准备。答案仅供参考，不一定是最优程序设计。

五、由于我们水平不高，这本书可能有不少错误，欢迎大家指正。

姜肇昆 赵志复 韩子敏 袁宜仲  
张丽 李 猛 王杰执笔

# 目 录

## 前 言

第一章	电子计算机的概述.....	(1)
第二章	BASIC语言和程序设计.....	(9)
一、	语言、程序.....	(9)
二、	BASIC 语言学习示意图.....	(14)
三、	运算符号.....	(16)
四、	数、变量、数学表达式.....	(16)
五、	运算规则.....	(19)
六、	NEW .....	(20)
七、	HOME (CLS) 清屏.....	(20)
八、	RUN 执行、运行.....	(20)
九、	LIST 打印程序清单.....	(20)
十、	PRINT 打印语句.....	(21)
十一、	LET 赋值语句.....	(25)
十二、	修改.....	(29)
十三、	FOR NEXT 循环语句.....	(30)
十四、	框图.....	(36)
十五、	REM 注释语句.....	(39)
十六、	TAB (X) 打印格式语句.....	(39)
十七、	INPUT 键盘输入语句.....	(41)
十八、	GOTO 无条件转向语句.....	(43)
十九、	IF THEN 条件判断语句.....	(44)

二十、子程序GOSUB转子语句RETURN	
返回语句·····	(48)
二十一、READ DATA读数置数语句·····	(52)
二十二、RESTORE 恢复数据区语句·····	(55)
二十三、函数(三角函数、开平方根函数、随机 函数、取整函数、绝对值函数、符号函 数、对数函数、指数函数、自定义函 数)·····	(56)
二十四、DIM 数组说明语句·····	(64)
二十五、ON 开关语句·····	(69)
二十六、逻辑判断·····	(71)
二十七、循环的转出·····	(73)
二十八、APPLE II型机彩色作图·····	(74)
二十九、LASER310机作图·····	(77)
三十、LASER310机的音乐程序·····	(80)
第三章 文件存取和打印机使用·····	(83)
第四章 指法训练·····	(87)
第五章 复习提要·····	(89)
附图一 APPLE II型机键盘图·····	92)
附图二 LASER310型机键盘图·····	(93)
APPLE II型机错误信息表·····	(94)
习题·····	(97)
习题参考答案·····	(114)
总复习练习程序·····	(140)
附图三 指法训练图示意·····	(164)

# 第一章 电子计算机的概述

## 一、电子计算机的发展概况

随着科学技术和生产的发展，于1946年世界上出现了第一台电子计算机“ENIAC”全机用了18000个电子管，6000个继电器，占地近170平方米，重约三十多吨，运算速度每秒钟可做5000次加，虽然用现在的眼光看，它有很多不足之处，但是电子计算机的诞生却不同于历史上任何机器的产生，它有条件的，部分的代替和解放了人的脑力劳动和体力劳动。从第一台电子计算机的诞生至今不足40年的历史，但它的发展却异常迅猛，今天已深入到科研、生产、军事、生活各个领域，1950年全世界只有25台电子计算机。到1970年就已有11万台。到1988年达175万台（不包括微型机）。

目前，电子计算机向两个方向发展，——巨型机和微型机。

大型机、巨型机主要用于科学计算、大型设计和大量数据处理等。

小型机、微型机主要用于企业管理，少量的数据计算，信息传送和控制等。微型机发展很快，每年差不多以100万台的数量进入市场。

我国的计算机事业是从1956年开始的，1959年研制成第一台电子管计算机，1983年又制成每秒运行亿次的巨型计算机。全国已拥有大、中、小型计算机中百分之八十都是我国

自己制造的。

党中央和国务院对计算机的生产和应用是非常重视的，强调指出：电子计算机是一种新技术，一种完全新型的生产力。从我国的国情出发，对微型机的利用，一定要摆在重要的位置予以足够的重视。现在微型机的使用范围非常广泛，就生产领域来说，它可以用于管理，用于生产过程的控制，用于技术改造，而且效果很明显。微型机的应用，对老企业的改造，特别是对中小企业的改造，将会起到我们预想不到的作用。计算机是现代化必不可少的工具。

计算机的作用不同于收录机、电视机，掌握电子计算机的人员知识越渊博，对电子计算机语言越精通，电子计算机也就能发挥出更大的效益，反之，就不能充分发挥出计算机的作用。因此，计算机普及要从娃娃做起，使青少年尽早地接触电子计算机，学习计算机知识和技术，是我们多年提倡和大力开拓的领域。

## 二、电子计算机有以下几个特点：

### （一）运算速度快。

电子计算机运算速度之快，是一般人力所做不到的，巨型机每秒钟运行几亿次，相当于一个人几十年上百年的计算工作量。

### （二）精确度高。

### （三）具有“记忆”、逻辑判断和选择的能力。

### （四）计算机内部的运算都是自动进行的。

### （五）通用性强。

## 三、电子计算机的用途极广，主要有以下几个方面：

### （一）数值计算，或称为科学计算。

(二) 数据处理和信息加工。

(三) 实时控制。

(四) 模拟智能。


电子计算机用途虽然非常广泛，但它毕竟是按人给它输入的程序工作的。人是计算机的主人，而不要变成计算机的仆人。

#### 四、电子计算机的组成及使用

各种电子计算机的设计目的不同，它的组成也有差异，这里着重介绍APPLE I微型电子计算机系统和LASER 310微型电子计算机的主要组成部分。

(一) APPLE I微型计算机(包括主机和外围设备)：

1. 主机：是计算机系统的核心部分。



中央处理器是电子计算机的核心，由它产生各种指令，并以此控制计算机的运行。CPU的振荡频率决定计算机的速度。

内存贮器是存放信息的，好象一个旅馆有许多“房间”，每个“房间”叫作一个存贮单元，每个单元有一个地址，就象房间号，每个房间中存放的数字的位数是固定的，八位微型计算机就是指每个内存单元能放一个二进制的八位数，这个数叫作这个单元的内容。内存贮器的容量是以K为单位的，1K就等于1024个存贮单元，APPLE I型微机内存有64K，就是

1024×64个单元,这64K是随机存贮的,关机后不再保留信息。还有12K监控,自动启动和BASIC扩展程序,在出厂前已存贮,关机后仍然保留信息。

接口是把主机与外部设备相联结的装置。

## 2.外部设备

### (1) 键盘 (见附图一)

通过键盘将编写好的程序输入主机。是人向计算机“说话”传达人的指示的工具。键盘上有53个键。

①数字键: 1、2、3、4、5、6、7、8、9、 $\emptyset$ 共十个。 $\emptyset$ 是数字零。

②英文字母键: 26个英文字母,与英文打字机字母排列位置相同。

### ③符号键与移位键。

在数字上方和一些键上有!、”、#、%、&、'、( )、^、\*、=、-、+、/、?、<、>、·、,、;、:、<-、->等符号。

当需用这些符号时,凡符号位置在按键上方的,必须先按住SHIFT移位键之后,再按符号,才能出现所需符号。

### ④RETURN回车键

在输入程序过程中,需要换行将光标移回到下一行的起始位置时(初学时在写程序中用符号↵表示↵,要按RETURN键。

### ⑤空格键 (SPACE)

在输入语句过程中需要留出空格位置时,要按空格键。

### ⑥RESET复位键

按复位键可使程序停止运行，恢复到执行前的状态。  
其它各键的功能、使用方法将在第二章中介绍。

### (2) 显示器

它是计算机向人报告它工作情况的设备，它可显示出输入的程序和运行的结果。显示器有彩色和绿（黑）白两种。

### (3) 磁盘驱动器

是计算机的外存贮设备。磁盘是用磁介质涂在塑料片上制成的。常用的有两种规格，即8吋和5¼吋。

磁盘是用来存贮程序的，磁盘上有35道同心圆，又分成16个扇区，全盘可存123K。

常用的外存贮器还有硬盘、磁带、卡片、纸带、录音机磁带等。


磁盘要求保持清洁，严禁用手接触小窗孔，用毕立即放入纸套中。存放磁盘的地方要防磁、防潮、防热。

### (4) 打印机

能打印出显示器上的程序和运行结果。是做文字资料保存、传递文字信息的设备。

## 3. 关机规则

电子计算机安装完毕后应检查电源电压是否稳定、正常。电源插座要按下图装置，插头中的火线不能插错。中线与地线之间应无电压。中线与地线之间的电阻不应超过4欧姆，不应把地线拴在自来水管上。



### (1) 开机

①将DOS3.3磁盘标号向上，缺口向左，慢慢送入驱动器内并轻轻关上驱动器的门；

②打开显示器的电源开关；

③打开主机电源开关，这时驱动器上红灯亮 并发出沙沙音，红灯没熄灭时不得移动磁盘；

④计算机系统开始自动调DOS3.3磁盘操作系统到主机内存贮器中。

┘ 这个符号叫提示符，是光标。提示符和光标出现后，驱动器红灯熄灭，沙沙音停止，计算机才可以工作。

如主机没连接驱动器，或没装入磁盘而又需要工作时，可按**RESET**键，在屏幕上即可出现提示符，可开始工作。

## (2) 关机

①在红灯不亮时打开驱动器门，取出磁盘装入专用纸套中；

②关闭主机电源；

③关闭显示器电源；

④登记使用机器的情况；


## (二) LASER310微型电子计算机

它的主机与键盘联在一起，也可用一般电视机做为计算机的显示器，可将主机输出线接在电视机的二频道上。

1. 键盘与使用（见附图二）共46个键。键的字符分别标在四个位置。如图：

(1) 数字键与英文字母键。与APPLE I微型计算机相同。在标1处。用时可直接按这个键。

(2) 符号键、图案键与移位键：符号键与APPLE I型机基本相同，但没有八号，乘方改用↑号 还有16个小图



案，可用小图案组成各种图形。设在标2处。用时需先按住移位键**SHIFT**，再按所需符号键或图案键。

(3) 指令键。在标3处有各种指令，如果不用键盘输入单个字符组成指令时，可以按**CTRL**键 (**CONTROL**)，同时按所需用的指令键。

(4) **RETURN**回车键、**SPACE**空格键，用法与**APPLE I**型机相同。

(5) 光标反白键。按**INVERSE**键后，光标反白，再按其它键输入字符时，屏幕上显示白底黑字。再按**INVERSE**键，就可以退出反白，恢复原状。

(6) 函数键。在标4处设有函数键。(但在键面上没有表示出来)。如果不用键输入单个字符组成函数时，可以按住**CTRL**键，再按一下**FUNCTION**键，再按所需用的函数键。

**LASER310**型机也可带驱动器、录音机、打印机等外部设备。

## 2. 开关机规则：

### (1) 开机：

① 打开显示器电源开关；

② 显示器屏幕亮后，打开主机电源开关，主机红色指示灯亮；

③在显示器屏幕左下角出现光标后，就可以开始工作。

、2) 关机：

- ①关闭主机电源开关；
- ②关闭显示器电源开关；
- ③登记使用机器情况。

五、在使用键盘向主机输入时，要注意指法。第四章讲指法训练。指法训练非常重要，应当在开始学习语言时，就同时进行指法训练。

## 第二章 BASIC语言和程序设计

### 一、语言、程序

从习题（97页）的第一、二、三题中，可以看出，计算机接受了人们向它发布的命令，按照人们的意图进行了工作。以这几道题为例，人们向计算机发布了什么命令呢？请将下面左右两栏对照来看吧！

第一题是人命令计算机计算一加一，再加一，……直加到一百个一的结果。

NEW	清除内存的程序
10 LET S=0	S是累加先给S置0
20 FOR I=1TO100	I是变量从1变到100
30 LET S=S+1	S每次加1
40 NEXT I	执行下一个I的量
50 PRINT "S=" ; S	打印S值
60 END	结束
RUN	执行

第二题是一百块正方形水泥板，最外块边长一米。每块依次边长增加一米，人命令计算机计算这一百块水泥板的面积总和。

NEW	清除内存的程序
10 LET S=0	S是面积累加和先给S置0
20 FOR X=1TO100	X是变量从1到100
30 LET S=S+X*X	S是变量X乘X累加和, X每 次增1
40 NEXT X	执行下一个X的量
50 PRINT "S=", S	打印S值
60 END	结束
RUN	执行

第三题是人命令计算机打印 \* 这样的图形

```

*
* * *
* * * * *

```

NEW	清除内存的程序
10 FOR I=1TO 10	变量 I 从1变到10
20 PRINT TAB (15-I) ;	打印格式第一个字符打印在 15- I 这列
30 FOR J=1TO 2*I-1	变量 I 从1变到2乘 I 减1
40 PRINT " * " ;	打印 * 号从左向右 连打不 换行
50 NEXT J	执行下一个 J 的量
60 PRINT	打印空行, 使 * 号换行打印
70 NEXT I	执行下一个 I 的量
80 END	结束
RUN	执行

在这里, 人们是用BASIC语言编制程序向计算机发布

了许多命令。

(一) 语言：计算机要按照人们的意图工作，就必须使计算机懂得人的意图，接受人向它发布的命令。但它不懂人说的语言，就需要人用计算机语言同它讲话、发令。

计算机语言有机器语言、汇编语言、高级语言。机器语言用二进制作代码，用起来很困难，难学，难写，难记，难检查，难修改，又不通用，于是人们又创造了高级语言。高级语言包括BASIC、FORTRAN、ALGOL、COBOL等。人们用这些高级语言给计算机发指令。计算机自己把这些语言翻译成机器指令程序去执行。人们学会了高级语言就可以很快地学会使用计算机，而且可以不管什么机器指令，也不必懂得计算机内部结构和工作原理，就可以使用计算机。高级语言还有一个好处，是可以适用于不同的计算机。

(二)、BASIC语言：BASIC语言是初学者通用符号指令代码的英文缩写(Beginner's All-Purpose Symbolic Instruction Code)，是一种小型会话式高级语言。但现在它已不只是初学者使用，而是一种国际通用的计算机语言了。小型机、微型机一般都配有BASIC语言。BASIC语言已经由简单的BASIC，发展成为扩展BASIC，我们学习的是APPLE I微型计算机的浮点扩展BASIC语言。

这种语言的特点：

1. 简单易学，基本BASIC语句只有17种（参见第15页BASIC语言学习示意图和第89页复习提要。符号同数学符号差不多，许多指令就同英语一样（如PRINT、END、NEW等）。

2. 会话式语句。可以人机对话，也便于查找程序中的错

误。如果程序有错误，执行后，屏幕上就会显示出SYNTAX ERROR的字样。如果执行到第40句时程序中断了，就会在屏幕上显示出BREAK IN 40字样。

3.有立即执行语句和迟缓执行语句两种方式。

(1)立即执行语句：前面没有行号，按回车键立即执行，只能执行一次。如

```
PRINT 3+4
```

7

(2)迟缓执行语句：有行号，在输入运行命令(RUN)后执行，可执行多次。如

```
10 PRINT 3+4
```

```
20 END
```

```
RUN
```

7

```
RUN
```

7


4.有数据处理能力。不只是做数学运算，而且包括逻辑判断、文字、信息、情报、编写文件等。

5.不同的计算机在使用BASIC语言时有小的区别，在使用时要查阅使用说明书，但它们基本通用。

(三)程序：人们做事都要有步骤、有进度、有次序。如今天大家来上课，从进楼门开始排程序，直到听课(见下页)。

这样表达出来的步骤就是程序。这个图形是处理问题时的思维图形，在计算机程序设计里叫框图，是执行的步骤。

计算机程序是由计算机语言来表达的，编制它的过程叫程序设计。编制程序就是用计算机语言编写一个让计算机做



某些事的步骤，让计算机按这些步骤办事。

#### (四) BASIC程序的构成和基本原则：

以第一章的几题为例，可以看出，BASIC语言程序有如下规则：

1. 由若干行组成；
2. 每一行叫一个语句 (APPLE II 型机和 LASER 310 型机允许在一行里写几个语句，各语句中间以符号：分开，称为复合语句)，但 APPLE II 型机每行不能超过 255 个字符，LASER 310 型机每行不能超过 64 个字符；
3. 每一个语句分别让计算机执行某个指令，做某一方面的事。

BASIC语言程序的样式一般有三部分：

1. 行号 (标号、语句标号)。如 10、20……行号必须是正整数。APPLE II 型机的行号可有 1~63999，LASER 310 型机的行号可有 1~65529。行号规定了计算机执行语

句的顺序，由小到大顺序执行，还可用来识别执行顺序的变化。键盘操作命令前不用行号。

每次改行号要另起一行。打完一行语句，要按回车键。

在排行号时一般按10、20、30、……排列，便于修改，便于在当中增加语句，如加入25、28句。如没有按行号由小到大顺序输入程序，在执行时，计算机可自动按由小到大的顺序执行。如两个语句先后用了同一行号，后句就把前句冲掉。每个行号在一个程序里必须是唯一的。

2. 语句部分。行号后面的部分叫语句部分。必须有语句指令（也叫语句定义符，它规定计算机执行某种功能）及语句体（即需要执行的具体内容）。

例如： NEW （是键盘操作命令，前面不用行号）

10 LET S=1

20 PRINT S

行号 语句指令 语句体

3. 结束语句END。END表示程序到此结束，让计算机停止工作。使用LASER310型机时，可不用END指令，但这不是规范化的程序。如果这个程序有子程序时，计算机就可能多执行一次。

在BASIC语言中，有暂停语句STOP让计算机执行到STOP语句时暂停执行，以便检查和修改程序。如要继续执行，可输入CONT指令（CONTACT）就可以继续执行了。

END指令同STOP指令不同的是，END不能继续执行，而STOP可使计算机显示出停止在那一行，可继续执行。

## 二、BASIC语言学习示意图

BASIC 语言学习示意图 (按 APPLE II 型机)

轮	打印作图	数据输入	算术运算函数	转移	条件判断	循环	文件存取	打印机的调用	系统调用	其他
1	PRINT	LET	+, -, *, /, <>, <=>, <>, <>, <>	GOTO	IF关系判断 THEN 行号	FOR NEXT DEL	LIST NEW DEL	PR #1 PR #0		END STOP REM
2	PRINT TAB( )	INPUT	函数 数组 DIM	GOSUB RETURN	IF关系判断 THEN 语句	多重循环	LOAD SAVE CATALOG DELETE	大小写	ITN FP PR #6	
3	彩色作图 GR COLOR PLOT X,Y VLIN Y <sub>1</sub> , Y <sub>2</sub> AT X HLIN X <sub>1</sub> , X <sub>2</sub> AT Y HGR HCOLOR HPLOT X <sub>1</sub> , Y <sub>1</sub> TO X <sub>2</sub> , Y <sub>2</sub>	READ DATA RESTORE	DEF FN	ON GOTO ON GOSUB ON ERR GOTO	IF逻辑判断 THEN 行号或语句	循环 转出	INIT HELLO COPY RUN		汇编	汉字

按照这个示意图学习，有以下特点：

1. 可以由浅入深分三轮进行。
2. 把理论学习、编制程序和上机操作结合起来，可使初学计算机的人在短时间里学会。
3. 对不同文化程度的学员可有不同的要求。

### 三、运算符号：

1. 代数式运算符号有：

加	减	乘	除	乘方
+	-	*	/	^ (LASER310机用^)

2. 关系式运算符号有：

等于	大于	小于	不等于
=	>	<	<>或><

小于等于	大于等于
<=	>=

3. 括弧：没有大、中、小之分，可以套用，但必须对称，不能失配。不能只有(,没有),或是只有),没有(。如：  
( ( ) )、 ( ( ) ) )

### 四、数、变量、数学表达式

(一) 数：

1. 在电子计算机里一般数的科学表示法是：

S	X.XXXXXXXXXX	E	S	TT
符号	0~9的九位数	以10为	符	1~9的二
“+”或“-”，		底的指	号	位数，最
“+”可不打上符号		数		多可到38

如：1.234E+19 表示 $1.234 \times 10^{19}$

5.876E-20 表示 $5.876 \times 10^{-20}$

要看， $10^{-3}$ 要写成1E-03，不能写成E-03。指数部分不能单独表示一个数。正负号不能省略。

2. APPLE II型机、LASER310型机实型数的范围是  
 $-1 \times 10^{38} \sim 1 \times 10^{38}$

(二) 变量：是允许改变它的值的量。

从以下几例，可以看出X值的变化：

```
1) 10 X=1
    20 PRINT "X=", X
    30 END
    RUN
    X=1
```

```
2) 10 FOR I=1 TO 5
    20 X=I
    30 PRINT "X=", X
    40 NEXT I
    50 END
    RUN
    X=1
    X=2
    X=3
    X=4
    X=5
```

```
3) 10 FOR I=1 TO 5
```

```

20 X = X + 1
30 PRINT "X = ", X
40 NEXT I
50 END
RUN
X = 1
X = 3
X = 6
X = 10
X = 15

```

变量（如以上三例中的X）是什么？从X值的变化可以看出：

1. 是量；
2. 值是不固定的，有的X值在执行程序的过程中可以变动；

3. 占一个存贮单元；

4. 变量的形式必须是英文字母开头，英文字母必须大写，第二位和以后可以是字母或数字字符，叫做变量名，如A、B、AC、A1、B5、BE576等。取什么变量名可以自定，通常使用便于识别并能代表变量特性的字符来命名。不允许用O、7B、2A等。用APPLE II型机时，变量名最多可有238个字符，但计算机只认前两位。LASER310机只允许两位。

5. BASIC语言中变量有简单变量和下标变量两种。下标变量留到以后再讲，简单变量又有实型变量和字符串变量。

实型变量是代表一个实型数的变量如  $A = 3 * 5$

字符串变量是可以表示文字，在一个英文字母后面加上表示符 \$。

如：A\$ = "JI SUAN JI" (计算机)

C\$ = "WANG PING" (王萍)

在字符串变量的 = 号右边的字符串必须加引号。

(三) 数学表达式：计算机工作是按照数学表达式进行的，计算机的数学表达式是按照计算机语言的规定，用运算符号、括弧等把数、变量、函数等连接起来的数学式。

如： $5x^2 - 3x - \frac{2 \sin a}{3}$  的数学表达式是：

$5 * X \wedge 2 - 3 * X - 2 * \text{SIN} (A) / 3$

又如： $\frac{a}{-b * c}$  的数学表达式只能写成：A / (-B \* C) 不能写成：A / -B \* C

## 五、运算规则

运算要遵循下面先后顺序：

(一) ( )。没有大、中、小之分，可以套用，但必须对称。

(二) 函数。

(三) 乘方。

(四) 乘、除。

(五) 加、减。

(六) 同级运算要遵照先左后右的规则。

要意注以下两对数学表达式中①式、②式是不同的：

①  $23 * A / 2 + B$  的算术式是  $\frac{23 A}{2} + B$

②  $23 * A / (2 + B)$  的算术式是  $\frac{23 A}{2 + B}$

①  $23 * A / 2 * B$  的算术式是  $\frac{23 A}{2} \times B$

②  $23 * A / (2 * B)$  的算术式是  $\frac{23 A}{2 B}$

**六、NEW** 这条指令是要清除主机内存程序。是 键盘操作命令，因此指令前不用行号。

在每次输入一个新程序前，必须先输入NEW指令，将主机内存的原有程序清除。否则新程序输入后，有些原有程序没被冲掉，将造成新旧交叉的混乱程序。

### 七、HOME (LASER310机用CLS) 清屏

这个指令是把整个屏幕上的字符清除，但是并没有清除主机内存程序。是键盘操作命令。

### 八、RUN 执行、运行

这个指令是叫程序运行。是键盘操作命令。

### 九、LIST 打印程序清单

执行这个指令可在显示器屏幕上显示出主机内存的程序。是键盘操作命令。

如显示器屏幕已满。需要暂停继续显示下面的程序时，可按RESET键。（用LASER310型机时，按空格键，就可以暂停，如要再继续显示时，可再按空格键）。

LISTn (行号) 这个指令是要在屏幕上显示出第n行语

句。如：LIST 90

LIST n1, n2(LASER310机是LIST n1-n2)

这个指令是在屏幕上显示n1行到n2行语句。

LIST120, 240 (LASER310机要输入LIST120-240)

## 十、PRINT 打印语句

在BASIC程序中，几乎每个程序都要用PRINT指令，进行打印。它的用法有以下几种：

(一) 打印数。如：

```
10 PRINT 3
```

```
20 PRINT
```

```
30 PRINT
```

```
40 PRINT 6
```

```
50 PRINT
```

```
60 PRINT 9
```

```
70 EDN
```

```
RUN
```

```
3
```

```
6
```

```
9
```

(二) 打印已被赋值的变量。如：

```
10 LET A = 3
```

```
20 PRINT A
```

```
30 END
```



```

10 LET A = 2
20 LET B = 4
30 LET C = 6
40 PRINT A, B, C
50 END
RUN

```

2                      4                      6

注：LASER310型机标准格式是按列分成两部分：0~15列，16~31列。LASER310机在屏幕显示时虽然正号（+）可省略不显示，但仍占一列的位置，因此数字要向右推一列显示。

#### （五）打印字符串。

用引号把某些字符括出来，称为字符串。编入程序，由计算机全部照原样打印起来。例如：

```

1) 10 PRINT " * * * "
    20 PRINT " $ $ $ "
    30 PRINT " 3 4 3 "
    40 PRINT "MHM"
    50 END
    RUN

```

\* \* \*  
\$ \$ \$  
3 4 3  
MHM

```

2) 10 PRINT " * * * * * * * * "
    20 PRINT " * * * * * * * "

```

```
30 PRINT "*" "
```

```
40 PRINT "*****"
```

```
50 END
```

```
RUN
```

```
*****
```

```
*          *
```

```
*          *
```

```
*****
```

```
3) 10 PRINT "A", "B", "C"
```

```
20 PRINT "A", "B", "C"
```

```
30 PRINT "A",
```

```
40 PRINT "B",
```

```
50 PRINT "C"
```

```
60 END
```

```
RUN
```

```
A          B          C
```

```
ABC
```

```
ABC
```

符号；的用法：符号；使前后字符从左到右连接起来打印。LASER310机用符号；打印数时，数左右各空一字符。

```
4) 10 LET X=3
```

```
20 PRINT "X=" ; X
```

```
30 END
```

```
RUN
```

```
X= 3
```

```
5) 10 PRINT "3 + 4 =" ; 3 + 4
20 END
RUN
```

3 + 4 = 7

这里要注意

1) PRINT X, Y, Z

与 PRINT "X", "Y", "Z"

与 PRINT "X"; "Y"; "Z" 的区别

2) 打印语句是不能用赋值的, 如:

```
10 PRINT A = (X + Y + Z) / Y
```

```
10 PRINT A = 2 * 4
```

都是错误语句。

3) APPLE II 型机、LASER310型机 PRINT 有方便键也叫功能键就是一个键可代表一个语句指令的功能。可用符号? 代替, 就是按住SHIFT键再按键? 可代替PRINT。

LASER310 型机有PRINT键。

## 十一、LET 赋值语句

LET语句是提供数据的语句之一, 是用来给程序中的变量赋值的, 可给变量赋予确定的数值或字符串。

形式: LET 变量名 = 数学表达式 (或字符串)

执行这个语句的结果是在计算机的内存里开一个存储单元, 存放变量, 并把数学表达式的结果, 赋给这个变量, 就是把 = 符号右边的值送给 = 符号左边的变量。好象开一个房间, 住进住户。符号 = 不是数学里的“等于”, 而是借用符号 =, 好象一座“桥”, 把河右边的住户过渡到左边的房间

里。这座“桥”是单行路桥，过这座“桥”的，只许从右往左走。

例题：1)

10 LET X=3	10 LET X=3
20 LET Y=4	20 LET Y=4
30 LET X=Y	30 LET Y=X
40 PRINT X, Y	40 PRINT X, Y
50 END	50 END
RUN	RUN
4            4	3            3

3)

10 LET X=3	10 A\$="TIAN JIN"
20 PRINT X	20 B\$="BASIC"
30 LET X=X+1	30 PRINT A\$, B\$
40 PRINT X	40 END
50 END	RUN
RUN	TIAN JIN BASIC
3	
4	

从以上例题可以看出符号=不是数学里的“等于” $X=X+1$ 在计算机程序里是正确的，而在代数式里则是错误的。不要同数学概念相混淆。

在输入指令时，LET可省略。

例题5)

```
5 A=0:B=0:C=0:D=0:E=0
10 B=12
```

```

20 C = A + B + 3
30 D = A + B - 4
40 E = A
50 PRINT A, B, C, D, E
60 A = 6
70 E = A
80 PRINT A, B, C, D, E
90 END

```

RUN

0	12	15
8	0	
6	12	15
8	6	

LASER310机执行结果是：

RUN

0	12
15	8
0	
6	12
15	8
6	

在执行这个程序时：各语句中各变量的变化情况如下。

	A	B	C	D	E
5	0	0	0	0	0
10	0	12	0	0	0
20	0	12	15	0	0
30	0	12	15	8	0
40	0	12	15	8	0
50	0	12	15	8	0
60	6	12	15	8	0
70	6	12	15	8	6
80	6	12	15	8	6

对于LET语句要说明几点:

1、一个变量赋值给其它变量后自己值不变。一个变量被赋值后,原值被新值代替,不再保留。

2、给字符串变量赋值时,字符串要用符号“ ”引起来。

3、变量如没赋值,就按0对待。

4、符号=两边如颠倒互换,语句的含义也改变了。如  $X = Y$  改成  $Y = X$  时则意义不同。

5、每个语句中不允许超过一个=符号。如:  
 $X = 3 * Y + 2 = (Y + 1) * (Y + 2)$  是错误语句,

6、不允许向表达式赋值。如:  $X + Y + Z = 1 + 2 + 3$  是错误语句。

## 十二、修改

APPLE I 型机修改方法有：

1. 改单个字符：将光标移在要修改的字符上，输入新字符就可将旧字符改掉。

2. 修改一行语句：输入这行语句的行号再按回车键，就可以删除这行语句。或输入行号后再输入新语句，就冲掉原行号语句。

3. 使用DEL n1, n2删除n1行号到n2行号之间的语句，重新LIST就能删除内存中的程序。是键盘操作命令。

4. 使用ESC键 (ESCAPE) 修改：在一个程序经过LIST列出清单以后使用ESC键。步骤是：

按一下ESC键；

按I键，使光标上移到要修改的语句行号处；

按J键，使光标左移到本行第一列；

按→右移键，将光标移到要修改的字符上进行修改；

修改后再用→右移键将光标移到本语句最后一个字符右边。但不要移到下一行语句上，以免将光标经过的下一行语句的字符，输入到本语句里。

按回车键，将光标移出本程序，再输入LIST指令，检查程序修改情况。

此外，按ESC键后，再按K键可让光标右移；按M键可将光标下移。

LASER301型机修改方法有：

1. 修改单个字符和修改一行语句：与APPLE I型机基

本相同。但在修改单个字符时可用RUBOUT键除掉原字符。

2. 当一个程序作过LIST列出清单后发现错误需要修改时，可分别按上、下、左、右四个箭头键，将光标移到所要修改的字符处，再用修改单个字符的办法修改。修改完毕，需要按RETURN键，才能在主机内存中变成新程序。

3. 如果需要在两个字符中间插入字符，而不改变原来的字符，可把光标移到要插入的位置后面一个字符上，再按INSERT键，再用键输入新的字符，就可以达到预想的结果。

### 十三、FOR NEXT 循环语句

是用于多次重复执行一条（或一组）语句的指令。本节只介绍两种循环。

（一）单层循环。从三个例题谈起：

例一是习题十六，如果只用PRINT语句，打印出那个图形是很麻烦的，我们可以设计一个程序，就比较省事：

```
10 FOR I=1 TO 10 STEP 1
20 PRINT "*" ;
30 NEXT I
40 PRINT
50 FOR I=1 TO 10 STEP 1
60 PRINT TAB (10) ; "*"
70 NEXT I
80 FOR I=11 TO 20 STEP 1
90 PRINT TAB (I) ; "*"

```

```
100 NEXT I
```

```
110 END
```

还可以加上：

```
25 FOR U=1 TO 300
```

```
26 NEXT U
```

(注：在LASER310型机第60语句改为PRINT TAB(9)；“\*”。第80语句改为FOR I=10 TO 19 STEP 1。第25语句改为FOR U=1 TO 100。

打印出图形好象一个小虫子在石头上很慢的爬，爬到了边上，很快地垂直掉下去，又很快地从斜坡上滚下去。在这个程序里用了循环语句。在这个程序里，第10至40语句是打印横着的十个符号\*；50至70语句打印竖着的十个符号\*；80至100语句是打印斜着的十个符号\*。增加的25、26两个语句是一个“空循环”是为了延长第10至第30语句的执行时间。

例二是要计算20! 就是20的阶乘： $1 \times 2 \times 3 \cdots \cdots$   
 $\times 20 = ?$

```
10 LET X=1
```

```
20 FOR I=1 TO 20 STEP 1
```

```
30 X=X*I
```

```
40 NEXT I
```

```
50 PRINT "X=" ; X
```

```
60 END
```

```
RUN
```

```
X=2.43290201E+18
```

这里也使用了循环语句，比用计算器速度要快得多。

例三是：相传古代印度国王舍罕要奖赏他聪明能干的宰相达依尔，国王问达依尔你要什么，达依尔说：我只要国王在国际象棋棋盘的第一个格里放一粒麦子，第二个格里放两粒，第三个格里四粒，以后依次每格加一倍。一直放到棋盘最后一格，第六十四格，把这些麦子奖给我，我就感恩不尽了！国王想：这还不容易！就叫人扛来一袋小麦，但不一会儿就用完了，再扛来一袋，很快又用完了，国王很奇怪，算不清这笔帐。我们用电子计算机来算一下，即求S（麦粒之和） $= 1 + 2 + 2^2 + 2^3 + \dots + 2^{63}$

10 S = 0 S是麦粒数之和，先置零。

20 I = 0 变量I是指数，指数从0开始即 $2^0$

30 FOR I = 0 TO 63 指数从零到63即  
 $2^0 + 2^1 + 2^2 + 2^3 + \dots + 2^{63}$

40 F = 2 ^ I

50 S = S + F

60 NEXT I

70 PRINT "S = " ; S

80 M = S / (1.42E + 08) 1立方米小麦约有  
 $1.42 \times 10^8$ 粒

90 PRNIT "M = " ; M

100 END

RUN

S = 1.4467441E + 19 (麦粒数)

M = 1.29906648E + 11 (立方米)

从以上三例可以看出循环语句的用处是很大的。形式：

FOR 循环变量 = 初值 TO 终值 STEP 步长

NEXT 循环变量

初值、终值、步长可以是数，也可以是已被赋值的变量。

说明：

1. 循环体：FOR到与之相应的NEXT两条语句当中的语句叫做一个循环体。如例三40、50语句。

2. 循环变量与循环体中语句有两种关系。

一是循环变量在循环体中语句里出现，是循环体里变量和数学式的关系。如例二的I。循环变量也是循环体里的变量，循环语句有给循环变量连续赋值的作用。

二是循环变量在循环体中其它语句里不出现，只是控制循环体里其它语句执行的次数。如例一10~30句的I，只起控制作用。

循环体被重复执行的次数，取决于FOR后面的循环变量。

STEP步长是1时，可省略，隐含在语句中，如STEP步长是0时，循环变量不变。STEP步长可以是负数。但必须符合初值到终值的步长规律。

如：FOR I=10 TO 1 STEP -1是正确的。

FOR I=1 TO 10 STEP -1是错误的。

4. 退出循环体。当循环变量越出终值时，就退出循环体。开始执行NEXT的下一行语句。

(二) 双重循环。在编制程序时，有不少问题仅靠单层循环是不能解决或比较麻烦，就要用双重循环，组成循环嵌

套。

例一，打印图形：

```
*****
*****
*****
*****
```

如用单层循环程序是：

```
10 FOR I=1 TO 4
20 PRINT "*****"
30 NEXT I
40 END
```

改用双重循环程序是：

```
10 FOR I=1 TO 4      变量I代表行数，从
                    1行到4行
20 FOR J=1 TO 9      变量J代表每行
30 PRINT "*",        "*"号个数，从1
40 NEXT J            个到9个
50 PRINT
60 NEXT I
70 END
```

10语句至60语句是外循环，20语句至40语句是内循环，当变量值是J时，变量J值要从1变到9，30语句要执行9次，连续横排打印出九个\*号。当J越出终值，退出内循环体，执行到50语句时，仍连接第九个\*号打一空行。然后在执行60语句时要另起一行，而变量I值从1变到4，每增加1，内循环就要再执行一次，J重新取新值，从1变到9，打印九个

\*号。当I增值越出终值4时，退出外循环，结束。

例二，打印图形：

```
*  
* *  
* * *  
* * * *
```

程序是：

```
10 FOR I=TO 4  
20 FOR J=1 TO I  
30 PRINT "*" ;  
40 NEXT J  
50 PRINT  
60 NEXT I  
70 END
```

说明：



1. 在双重循环里每重循环的FOR变量和NEXT变量必须相对应。

2. 循环嵌套不得交叉。如：

正 确	错 误
FOR I	FOR I
FOR J	FOR J
⋮	⋮
NEXT J	NEXT I
NEXT I	NEXT J

#### 十四、框图（流程图）

常用的框有：



流程。反向时要用箭头表示。


一个框表示一段程序的功能。

运用框图的好处是使程序结构清晰，一目了然，形象直观；便于检查修改；使用程序人员之间的交流设计思想。所以画框图是编制计算机程序的基本功。

下面我们以第十三节中，单层循环例二来说明框图的作用：

由于程序当中有循环语句以及GOTO（无条件转向语句）、IF-THNE（条件语句）和GOSUB（转于语句）语句，使得有些程序在执行时不完全是按行号大小顺序执行，往往容易搞混程序走向。在编制程序时，特别是一些较长、较复杂的程序，也往往需要先借助一种图形来理清设计程序的思路，这种图形就叫做框图，也叫做流程图。

打印20!




以第十三节双重循环的例一说明框图的使用:

打印

```
*****  
*****  
*****  
*****
```

的框图是



框图的使用步骤是：

1. 确定计算方案；

2. 画框图并检验无误；要有依据，从核心部分入手，再向两端发展，逐步完善；要同语句对应，再经检验修定；修改核实。

3. 参考框图编制程序；

4. 调试运行程序，如有错误可对照框图检查。

## 十五、REM 注释语句

形式：

```
70 REM NAME
```

```
70 REM 20!
```

注释语句在程序里起注释作用，它是为了使自己和别人能够看清楚程序（或一部分程序）的内容。如：说明程序名称、计算方法、目的、功能、步骤等。当计算机执行LIST指令时REM语句中的全部字符都在屏幕上照原样显示出来，可以起到提示、备忘的作用，但REM语句不影响计算机其他语句的执行，一个程序可以有一个或几个REM语句，REM语句后面的字符不要用引号引起。

## 十六、TAB(X) 打印格式语句

TAB(X)是打印格式语句，可将字符打印在X列上，用来打印表格或打印图形。这是因为只打印标准格式，不能满足打印表格的需要，所以要使用打印格式语句。

形式：

```
PRINT TAB(X), A
```

X是数、数学式，代表第几列，APPLE I型机是1~40（列）之间的数（打印机可打1~80列），LASER310机是0~31（列）。大于这个数时，即自动换到第二行，继续打印。

A可以是数、赋值的变量、数学表达式或字符串。  
符号；处必须用；号。

```
例如：10 PRINT TAB (5) , " * "
        20 PRINT TAB (10) , " * "
        30 PRINT TAB (20) , " * "
        RUN
```

```
          *
          *
          *
10 PRINT TAB (5) , " * " ;
20 PRINT TAB (10) , " * " ;
30 PRINT TAB (20) , " * "
        RUN
          *          *          *
```

说明：

1.X指的是开始位置（列），包括负号。（LASER310型机正号也占一个位置）如：

```
10 A = 12
20 B = -12
30 PRINT TAB (5) , A
40 PRINT TAB (5) , B
```

```
50 PRINT TAB (5) , "12"
```

```
60 END
```

```
RUN
```

APPLE II 型机

	1	2	3	4	5	6	7	8
1					1	2		
2				-	1	2		
3					1	2		

LASER310型机

	0	1	2	3	4	5	6	7	8
1							1	2	
2					-	1	2		
3								1	2

2. X必须有确切值, 否则作零对待; X不能是负数; X如有小数, 即自动取整(如2.4即取2)。

3. 一个语句使用多个TAB时, X值只能从左到右顺序增加, 不能减少。请看下面正确和错误两种写法:

正确的: PRINT TAB (5) ; "A" ; TAB (10) ; "B" ; TAB (20) ; "C"

错误的: PRINT TAB (20) ; "A" ; TAB (10) ; "B" ; TAB (5) ; "C"

## 十七、INPUT 键盘输入语句

INPUT 语句也是提供数据的语句之一, 可以在程序运行后, 用键盘把数值输入给相应的变量。可给变量赋不同的值, 也可以给几个变量赋值。是人和计算机对话的输入语句。

形式:

① INPUT X

② INPUT X, Y, Z

③ INPUT "字符" ; X

#### ④ INPUT A \$

注：X、Y、Z代表数字变量，A \$是字符串变量。

说明：

1. 计算机执行到INPUT语句时，屏幕上出现一个？号，再由计算机使用者用键盘输入数据或字符串，输入后按回车键，把这个数值或字符串赋给相应的变量，然后计算机就连续执行下面的语句。如输入的数据或字符串有错误，计算机就不能继续执行下面的语句，显示器屏幕上就再次显示符号？直到正确输入数值或字符串。

2. 在运行完一次程序后，如需要键盘输入另外数值或字符串时，不需要写新程序，只要使程序重新运行一次，出现符号？时，再输入另外数值或字符串。

3. 输入数值必须是常数，不允许是函数或表达式。输入字符串时不需用符号“ ”引起来。

4. 用一个INPUT语句需输入若干变量值时，各变量中间用符号'分开，各数值中间也要用符号'分开。如果输入数值不足变量数时，显示器上就仍会显示出符号？，须继续输入数值。

例如：要编一个程序给全班50名学生分别计算语文、数学、英语、自然等四门课程的平均成绩，并在每人学号后打印出来。

3	INPUT N	输入全班学生人数
5	FOR I=1 TO N	I是学号
10	P=0	平均成绩P先置0
20	INPUT "NAME" ; M \$	输入姓名
30	INPUT "Y=" ; Y	输入语文分

```

40 INPUT "S=" ; S      输入数学分
50 INPUT "E=" ; E      输入英语分
60 INPUT "Z=" ; Z      输入自然分
70 P = (Y + S + E + Z) / 4
80 PRINT "NO." ; I, M $, P
90 NEXT I
100 END

```

每循环运行一次，即用键盘输入一个学生的姓名和四门课分数，分别计算打印。如果还要打印出各科成绩，怎么编程序请大家自己去想。

### 十八、GOTO 无条件转向语句

GOTO语句是用来无条件地改变程序的执行顺序，好象是一个指路标牌，指示程序的走向。

形式：

GOTO n            n是行号

如果要打印

```

* * * * *
$ $ $ $ $
# # # # #
* * * * *
$ $ $ $ $
# # # # #
* * * * *
$ $ $ $ $
# # # # #

```

```

* * * * *
$ $ $ $ $
# # # # #
: : : : :
: : : : :

```

可编程序：

```

10 PRINT " * * * * *"
20 PRINT " $ $ $ $ $"
30 PRINT " # # # # #"
40 GOTO 10
50 END

```

可连续执行。

### 十九、IF THEN 条件判断语句

在第三十二习题里，小明想用“\*”号打印两条交叉斜线，但他编制的程序是：

```

10 FOR A=1 TO 21
20 PRINT TAB (5+A) ; " * " ; TAB (26-
A) ; " * "
30 NEXT A
40 END

```

执行这个程序不能出现预想的结果，这是因为在程序执行过程中， $(5+A)$  的值会逐渐增大。 $(26-A)$  的值会逐渐减少。当 $(5+A)$  大于 $(26-A)$  时，第20语句就违反了打印格式语句的规定，会改变斜线的走向。

要想得到预定的结果，就要在 $(5+A)$  大于 $(26-A)$

时把第20语句变成PRINT TAB (26 - A) ; “\*” ; TAB (5 + A) ; “\*” ,先打印 (26 - A) 这列的\*号。这个程序可以使用IF THEN语句。即,

```
10 FOR A=1 TO 21
15 IF (5 + A) > (26 - A) THEN 25
20 PRINT TAB (5 + A) ; “*” ; TAB
    (26 - A) ; “*”
22 GOTO 30
25 PRINT TAB (26 - A) ; “*” ; TAB
    (5 + A) ; “*”
30 NEXT A
40 END
```


就是增加了第15、22、25三个语句。

第15语句起了判断的作用。

判断语句是有条件转向语句。

形式: IF 关系判断式 THEN n n是行号

这条语句的意思是,如果关系判断式的条件成立,就转



执行本行语句的下一行语句。

移执行第n行语句，如果条件不成立，就继续执行本行语句的下一行语句。

在使用框图设计条件判断语句时，要用判断框。

关系判断式可以是：

(一) 数值比较

1. 一个变量与另一个变量比较，如  $X > Y$ ；
2. 一个变量与一个数比较，如  $X < 5$ ；
3. 一个变量与一个数学表达式比较， $X > = A * 5$ ；
4. 两个表达式比较，如  $5 * A * Y < B * 8$ 。

(二) 字符串比较


1. 一个字符串变量与另一个字符串变量比较，如  $A \$ = B \$$ ；
2. 一个字符串变量与一个字符串比较，如  $A \$ = "TD-K"$

例一，列举一批数，打印出大于零的数并先画框图。

```
10 INPUT N
20 IF N > 0 THEN 40
30 GOTO 10
40 PRINT N
50 GOTO 10
60 END
```

或者：

```
10 INPUT N
20 IF N < = 0 THEN 10
30 PRINT N: GOTO 10
50 END
```



例二，以下这个程序，在什么条件下能打印出 \$\$\$\$

图形? 又在什么条件下能打印出 &&&& 图形? 请输入计算机检验。

```
10 INPUT "A $=" , A $
20 IF A $ = "ABC" THEN 50
30 IF A $ = "XYZ" THEN 60
40 GOTO 10
50 PRINT "$$$$" : GOTO 10
60 PRINT "&&&&" : GOTO 10
70 END
```

在键盘输入字符串是 ABC 条件下, 能打印出 \$\$\$\$ 图形, 在字符串是 XYZ 条件下能打印出 &&&& 图形。

例三, 用条件判断语句打印以下图形,

```
*
* *
* * *
* * * *
10 FOR I=1 TO 4
20 A=I
30 PRINT "*" ;
40 A=A-1
50 IF A=0 THEN 70
60 GOTO 30
70 PRINT
80 NEXT I
90 END
```

例四, 计算 20! 并画框图

10 I=0

I是变量表示阶乘数

20 X=1

X是变量表示阶乘的积

30 I=I+1

每执行一次增值

40 X=X\*I

X乘以变量I

50 IF I<20 THEN 30 I小于20时要转向30

语句使变量I和X增值

60 PRINT "X=" ;X

70 END

## 二十、子程序

**GOSUB**转语句

**RETURN**返回语句


在一个程序里，如果要多次进行某种运算或多次执行某个、某几个语句，就要重复书写这些语句，很烦琐，也容易出错，就可以运用子程序的方法。

形式：**GOSUB n**

**RETURN**

n是行号

如有一半径10



厘米的圆板，上有半径3厘米、2厘米、1厘米的圆孔，求这圆板完整时面积和三个孔的面积各多少？

用子程序编制程序，可设X是圆板面积，S1是完整的圆板面积，S2、S3、S4是三个孔面积，R是半径。

```
主程序 {
10 R=10
20 GOSUB 500
30 PRINT "S1="; S
40 X=S
50 R=3
60 GOSUB 500
70 PRINT "S2="; S
80 X=X-S
90 R=2
100 GOSUB 500
110 PRINT "S3="; S
120 X=X-S
130 R=1
140 GOSUB 500
150 PRINT "S4="; S
160 X=X-S
170 PRINT "X="; X
180 END

子程序 {
500 S=3.14159*R^2
510 RETURN
```

说明；

1. 把需要多次重复使用的一组语句分离出来，编写成子程序，需要时再反复调用它，这个程序就是子程序，如上题500~510语句；调用子程序的程序就是主程序，如上题10~180语句。


2. 主程序调子程序时用GOSUB语句指令。子程序行必须排在主程序结束符行号以后，但行号不必与主程序行相连。在结束子程序时要用RETURN语句指令。

3. 子程序执行到RETURN语句指令时，就转回到GOSUB语句指令调用子程序那行语句下面的语句，继续执行GOSUB与RETURN必须对应使用。

4. 主程序不允许从子程序中间插入子程序。一个子程序只能有一个入口，但可以有多个出口，如子程序里有几个IF语句，就有多个出口。

5. 调用子程序的次数没有限制。

6. 子程序可以嵌套子程序。不同机型对能嵌套几重子程序的数量有不同限制。如：



例一，打印下面这个图形，要编制带有子程序的程序。

```
      *
     ***
    *****
   *****
  *****
 *****
*****
```

```
10 FOR I=0 TO 2
20 GOSUB 500
30 NEXT I
40 FOR I=1 TO 0 STEP -1
50 GOSUB 500
60 NEXT I
70 END
500 FOR J=-I TO I
510 PRINT TAB (10-I) ; " * " ;
520 NEXT J
530 PRINT
540 RETURN
```

例二，求 $S=A!+B!+C!$

A、B、C、用键盘输入

```
10 INPUT A
20 P=A
30 GOSUB 500
40 X=R
50 INPUT B
60 P=B
70 GOSUB 500
```

R是阶乘积

```

80 Y = R
90 INPUT C
100 P = C
110 GOSUB 500
120 Z = R
130 S = X + Y + Z
140 PRINT "S=" ; A ; " + " ; B ; " + " ;
      C ; " = " ; S
150 END
500 R = 1
510 FOR I = 1 TO P
520 R = R * I
530 NEXT I
540 RETURN

```

## 二十一、READ DATA 读数量数语句

先看一个程序：

```

10 A = 35
20 B = 37
30 C = 41
40 D = 58
50 E = 80
60 S = A + B + C + D + E
70 PRINT S
80 END

```

如果变量和数据再多，就会很麻烦，也容易出错，所以在需要给大量的变量赋值时，就可以用READ DATA 语

句解决这个问题。如把上一程序改成：

```
10 READ A, B, C, D, E
20 S = A + B + C + D + E
30 PRINT S
40 DATA 35, 37, 41, 58, 80
50 END
```

形式： READ 变量

⋮  
⋮

DATA

说明：

1. READ是执行语句，而DATA是非执行语句，只是为给READ语句准备数据或字符串，执行READ语句时，从READ语句中顺序取出数据或字符串，依次给READ语句中变量赋值。

2. READ语句中的变量可以有字符串变量；在DATA语句中必须有相应的字符串。

3. 变量、字符串变量和数据、字符串的顺序要相对应。变量间和数据、字符串间以‘号分开，在最后一个变量和数据、字符串后不加‘号。

4. DATA语句中数据个数要等于或大于READ语句中变量的个数。

5. DATA数据表有一个“暗指针”，每读完一数，指针就右移至下一个数据。再读数时就读这个数据。

6. DATA可放在程序内任何位置，不一定紧靠READ

语句之后。

7. DATA只能置常数和字符串,字符串要用符号“ ”引起来,不能置变量或表达式。

8.一个READ和一个DATA语句,可以分成两个或两个以上的READ和DATA语句,如前面那个程序可写成:

```
10 READ A, B
20 READ C, D, E
30 S = A + B + C + D + E
40 PRINT S
50 DATA 35
60 DATA 37, 41, 58
70 DATA 80
80 END
```

例题,求下列几组数的平均值:

- ① 35, 37, 41, 58
- ② 62, 71, 81, 95, 108
- ③ 256, 301, 203

```
10 INPUT N          N是每组数的数据个数
20 S = 0
30 FOR I = 1 TO N
40 READ A
50 S = S + A
60 NEXT I
70 Y = S / N
80 DATA             (分别置入①、②、③组数)
90 PRINT "Y = ", Y
```

100 END

在求第一组数平均值时，将4用键盘输入N，DATA后置第一组数。求第二组、第三组数平均值时，分别将5、3用键盘输入N，DATA后分别置第二、三组数。

## 二十二、RESTORE 恢复数据区语句

形式：RESTORE

执行这个语句时，计算机把数据表中的“暗指针”移回到数据表开头，使DATA语句中数据从头开始，但不是使程序回到第一个READ语句，也不是从READ语句第一变量重新读起。使用这个语句可以压缩程序。

例如：

```
10 READ A, B, D, C, E
20 PRINT A, B, C, D, E
30 RESTORE
40 READ F, G
50 RESTORE
60 READ I, J, M, N
70 PRINT F, G, I, J, M, N
80 DATA 30, 45, 60, 80, 100
90 END
RUN
30 45 60
30 100
30 45 30
45 60 80
```

### 二十三、函数

在BASIC语句中还有一些标准函数，可以直接引用。  
例如：三角函数，开平方根函数，随机函数，取整函数等等。现分别介绍如下：

#### (一) 正弦函数、余弦函数、正切函数

形式：SIN(X)


COS(X)

TAN(X)

X是数学表达式、变量。是弧度。

由度换算成弧度的方法是：角度 \* 3.14156/180

例题：雷达发出电波后1/3750秒收到飞机回波，已知雷达的仰角是 $22^\circ$ ，电波速度是 $3 \times 10^8$ 米/秒，问飞机高度是多少米？



$$S = 1/3750 \times 1/2 \times 3 \times 10^8$$

$$H = S \times \sin(22 \times \pi/180)$$

编写程序：

```
10 S=1/3750*1/2*3*10^8
20 H=S*SIN(22*3.14159/180)
30 PRINT "H=" ; H
40 END
```

## (二) 反正切函数

形式：ATN(X)

X是数值、数学表达式或变量。

得到的值是弧度，求角度数时应注意换算。

例如，求反正切函数1是多少度？

```
PRINT ATN(1)/(3.14159/180)
45
```

## (三) 开平方根函数

形式：SQR (X)

X为数值、数学表达式或变量。X $\geq$ 0，不能用负数。

例一，PRINT SQR(36)

6

例二，解  $ax^2 + bx + c = 0$

由一元二次方程求根公式知道：

$$X_{1,2} = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}$$

编写程序：

```
10 INPUT A
20 INPUT B
```

```

30 INPUT C
40 Z = B^2 - 4*A*C
50 IF Z < 0 THEN 120
60 Z = SQR(Z)
70 X1 = (-B + Z)/(2*A)
80 X2 = (-B - Z)/(2*A)
90 PRINT "X1=" ; X1
100 PRINT "X2=" ; X2
110 GOTO 130
120 PRINT "WU SHI GEN" (无实根)
130 END

```

#### (四) 随机函数

随机函数是开机后，有一振荡源。经分频后，随机函数发生器产生很多组数，叫随机函数，这些数没有一定规律，是人们不能掌握的数。

形式：RND(X) X是变数，但不起实数作用。

说明：

1. 在  $X > 0$  时每次得到一个新的随机数。
2. 在  $X \leq 0$  时每次得到一个与前次相同的随机数。
3. 随机数的值都在大于0小于1之间。

LASER 310机如果X是0，就产生0到1之间的随机数；如果X是正数，就产生1到X之间的随机整数，X不可是负值。

例一，打印不同的随机数。

```
10 PRINT RND(10)
```

```
20 GOTO 10
```

```
30 END
```

例二，在屏幕上打印出无规律的满天星（可用\*号）。

```
10 FOR I=2 TO 20
```

```
20 G=RND(5)*40
```

```
30 PRINT TAB(G); "*"
```

```
40 NEXT I
```

```
50 END
```

注：LASER310机第20语句改为G=RND(31)。

#### (五)取整函数

形式：INT(X)

X是算术表达式或变量。

说明：

1.取整函数得到的数值是小于或等于X值的最大整数。

例一， PRINT INT(3.2)

3

例二， PRINT INT(-3.2)

-4

例三， 10 INPUT A

20 INPUT B

30 X=INT(A+B)

40 PRINT "A+B=" ; A+B

50 PRINT "INT(A+B)=" ; X

60 GOTO 10

70 END

RUN

? 3.2

? 5.3

A + B = 8.5

INT(A + B) = 8

? -3.2

? -5.3

A + B = -8.5

INT(A + B) = -9

?

2. 取整函数，只取整数，如要小数“四舍五入”可在 INT 后面符号 ( ) 里的数值上加上 0.5。

例一，PRINT INT(9.6 + 0.5)

10

例二，PRINT INT(9.4 + 0.5)

9

3. 如果要在小数点后保留两位（例如计算人民币的元、角、分，只保留到分），可以用下面的办法：

10 A = 115.6364

20 X = INT(A \* 100 + 0.5) / 100

30 PRINT X

40 END

RUN

115.64

4. 如果要在小数点后保留到三位或五位，用什么办法。请大家自己去想。

5.把取整函数和随机函数联合使用可以给小学生出一些数学题,给小学生出几道每次数值不同的二位以内正整数加法题,如果答错了,可以再答,答对了,就再出一道题,程序是:

```
10 A=INT(RND(5)*100)
20 B=INT(RND(5)*100)
30 PRINT A, "+", B, "=?"
40 INPUT X
50 C=A+B
60 IF X<>C THEN 40
70 PRINT "GOOD!"
80 GOTO 10
90 END
```

#### (六) 绝对值函数

形式:  $ABS(X)$   $X$ 可是数、数学表达式或是变量,如同初等代数中求 $|X|$ 的值一样。

例如: PRINT ABS(-3.2)

3.2

#### (七) 符号函数

形式:  $SGN(X)$

当  $X > 0$  时  $SGN(2) = 1$

当  $X < 0$  时  $SGN(-2) = -1$

当  $X = 0$  时  $SGN(0) = 0$

### (八) 对数函数

形式: LOG(X)

指X的自然对数。

例如: LOG(5)就是求以e为底的对数 $\ln 5$ 。

$$e = 2.71828$$

以e为底的对数和以10为底的对数换算关系是:

$$\text{LOG}_{10} X = \frac{\text{LOG}_e X}{\text{LOG}_e 10}$$

### (九) 指数函数

形式: EXP (X)

以e为底的X次方。

例如, EXP(2)就是求 $e^2 = ?$

在APPLE II型、LASER310型机上, X小于88, 否则就要溢出。

### (十) 自定义函数

在BASIC语言中, 除应用前面讲到的标准函数外, 还允许使用者自己定义函数, 为我们结合本身情况编写程序提供了方便。

形式: DEF FN函数名 (实型变量) = 数学表达式 1

FN函数名 (数学式2)

函数名可任选用26个英文字母中任意一个。

括弧中的实型变量, 是一个虚设的变量, 代表所定义函数的自变量。

括弧中的数学式2可以是常数或是表达式, 也可以是已

赋值的变量。

数学式 1 是自己定义的函数形式。这就是用 FN 函数名 (实型变量) 这个语句指令来代表一个较复杂的数学表达式。

自定义函数内可以套用自定义函数, APPLE II 型机可以套 18 层。

例一, 做一个实验, 发现有以下规律:

$$y = 3x^2 + 6x + 5$$

设  $x$  分别等于 2、4、6、8、10、……20 求出不同  $x$  值对应的  $y$  值。

编写出程序是:

```
10 DEF FN Y(X) = 3 * X^2 + 6 * X + 5
20 FOR X = 2 TO 20 STEP 2
30 PRINT X, FN Y(X)
40 NEXT X
50 END
```

例二, 打印出半径分别是 1、2、3、4、5、6、7、8、9、10、22、24、26、28、30、32、34、36、38、40、55、60、65、70、75、80、85、90、95、100 厘米时的 30 个圆的面积。

程序是:

```
10 DEF FN W(R) = 3.14159 * R^2
20 FOR R = 1 TO 10
30 PRINT "R=" ; R, "W=" ; FN W(R)
40 NEXT R
50 FOR R = 22 TO 40 STEP 2
60 PRINT "R=" ; R, "W=" ; FN W(R)
70 NEXT R
```

```

80 FOR R=55 TO 100 STEP 5
90 PRINT"R="; R, "W="; FN W(R)
100 NEXT R
110 END

```

例三，当变量分别是2和3时，求两个自定义函数的和。

```

10 DEF FN A(X)=X+1
20 DEF FN B(X)=FN A(X)*2
30 PRINT FN A(2)+FN B(3)
40 END
RUN

```

11

注：LASER310型机中没有自定义函数。LASER310型机有一些函数键。

#### 二十四、DIM数组说明语句。

DIM是数组说明语句的语句指令。执行DIM语句，在计算机内存中开辟预定的存储单元，以留给数组的各元素存放数值。一维数组可存放7248个元素，多维数组受内存的限制。

##### (一) 下标变量

形式： $A(X)$   $A(X, Y)$

$X, Y$ 是下标，是0或正整数，也可以是有确切值的简单变量或数学表达式。

这是一种可以使程序简化的变量。

例如，求五个没有规律数的和（不能用循环语句，只能用读数组数语句）的程序是：

```

10 READ A1, A2, A3, A4, A5
20 H = A1 + A2 + A3 + A4 + A5
30 PRINT "H=" ; H
40 DATA 2, 5, 6, 3.1, 8.7
50 END

```

如改用下标数，程序是：

```

10 H = 0
20 FOR I = 1 TO 5
30 READ A(I)
40 H = H + A(I)
50 NEXT I
60 PRINT "H=" ; H
70 DATA 2, 5, 6, 3.1, 8.7
80 END

```

(二) 数组，按一定次序排列，作为一个整体来处理的一组变量，叫做一个数组。数组中的每一个变量叫做一个下标变量，同一个字母组成的那些下标变量，称为数组，好象一个小队。如  $A(1)$ 、 $A(2)$ 、 $A(3)$ ……等，叫做  $A$  数组。下标变量叫做数组元素，好象一个个队员。每个下标量（数组元素）在计算机里占一个存储单元。

有一个下标的数组，称为一维数组，如  $A(I)$ ；有两个下标的数组称为二维数组，如  $A(I, J)$ 。

(三) 一维数组说明语句。

形式：DIM 变量（下标）

前面那道例题求五个没有规律的数的和，程序还比较简单。如果要改成求100个没有规律的数的和，就十分麻烦，但

用一维数组定义语句设计，就比较简单，程序是：

```
5 DIM A(100)
10 H = 0
20 FOR I=1 TO 100
30 READ A(I)
40 H = H + A(I)
50 NEXT I
60 PRINT "H=" , H
70 DATA 2,5, 6, 3.1,8.7,.....(一百个数)
80 END
```

说明：

1.在执行到数组说明语句时，计算机就给数组留出一定的存贮单元，好象是叫某小队全体队员集合。如在执行到第5语句时，计算机就给A数组留了101个存贮单元（下标是从0开始到100），因此，数组说明语句必须放在数组元素使用之前。

2.数组元素少于11个时可不定义，即在程序当中如果没有数组说明语句时，就被认为是DIM变量(10)。数组下标的数值必须是0或正整数，如果是小数，就自动取整。

3.在同一程序中需要使用几个数组时，可以同时写明几个数组，中间以，号分开 如DIM A(30)，B(20)，E(15)。已用作下标变量的变量名，在同一程序中不要再用来作简单变量的变量名。

4.下标变量可以是数，可以是已赋值的简单变量，可以是算术表达式，也可以是下标变量，如A(B(4))，还可以以字符串的形式出现，如要打印全班40个学生的姓名可以

编程序：

```
10 DIM A$(40)
20 FOR I=1 TO 40
30 READ A$(I)
40 PRINT A$(I)
50 NEXT I
60 DATA "LI HONG", "WU PING",
      .....(四十个人姓名)
70 END
```

如果要用标准格式打印姓名、语文成绩、数学成绩，可用三个数组说明语句：

```
10 DIM A$(40),Y(40),S(40)
20 FOR I=1 TO 40
30 READ A$(I),Y(I),S(I)
40 PRINT A$(I),Y(I),S(I)
50 NEXT I
60 DATA "LI HONG",97,99,"WU PING",
      96,100,.....(四十个人姓名及成绩)
70 END
```

(四)、二维数组说明语句。

带有两个下标的变量，称为双下标变量，如 $A(X, Y)$ ，由双下标变量组成的数组，叫做二维数组，二维数组说明语句表明所用数组是二维的。它可以得到有行和列的表格，一个下标表示行，一个下标表示列。

形式： DIM 变量(下标, 下标)

例如：某学校各年级、各班学生人数如下表：

年级/班级	1班	2班	3班	4班
1年级	36	40	38	39
2年级	38	37	39	42
3年级	33	35	34	33
4年级	39	32	33	30
5年级	39	37	35	0
6年级	35	37	37	0

当讲到各年级各班人数时，我们可以说一年一班是36人  
 一年二班是40人  
 一年三班是38人  
 一年四班是39人

为简便起见，也可改写成：

36(1,1) 40(1,2) 38(1,3) 39(1,4)。全校各班  
 都可以这样写出来。

如果要求打印出各班、各年级和全校的总人数，可用二维数组说明语句编写如下程序：

```

10   G = 0                               G是全校总人数
20   DIM A(6,4)
30   FOR I = 1 TO 6
40   H = 0                               H是年级人数
50   FOR J = 1 TO 4
60   READ A(I,J)
70   PHINT TAB(8 * J), A(I,J),
80   H = H + A(I,J)
90   NEXT J
100  PRINT

```

```

110 PRINT H
120 G = G + H
130 NEXT I
140 PRINT G
150 DATA 36,40,38,39,38,37,39,42,36,
        35,34,33,39,32,33,30
160 DATA 39,37,35,0,35,37,37,0
170 END

```

## 二十五、ON 开关语句

开关语句又称为控制转向语句。LASER310 机没有这个语句。

形式：ON I {GOTO } L1,L2,L3,……Ln  
           {GOSUB }

I是算术表达式或有确切值的变量，但必须大于0，如小于1，就按1对待；如有小数，就自动取整。

L1,L2,L3,……是行号

当I是1时转向第一个行号的语句，是2时则转向第二个行号的语句。……当I等于n时转向第Ln个的行号语句，但不能大于n。I好象是一把钥匙，去开Ln房间的锁，进入这个房间。

例：某商店售出各商品如下表，求出各种商品和总销售金额。

货物名	编号	售出量	单价	金额
毛巾	1	102	0.53元	?
牙膏	2	268	0.62元	?

茶杯	3	76	3.25元	?
饭盒	4	42	1.88元	?
录音机	5	65	306.50元	?
⋮	⋮	⋮	⋮	
⋮	⋮	⋮	⋮	

总金额?

设 M是各种货物售出量

G是各种货物金额

H是全部货物金额

编写程序如下:

```

10  INPUT I, M
20  ON I GOSUB 100,200,300,400,500, .....
30  PRINT G
40  H = H + G
50  PRINT H
60  GOTO 10
70  END

100 G = M*0.53
110 RETURN
200 G = M*0.62
210 RETURN
300 G = M*3.25
310 RETURN
400 G = M*1.88
410 RETURN
500 G = M*306.5

```

## 510 RETURN

⋮  
⋮

开关语句的另一种用法是在程序中出现意想不到的错误时，不使机器停止运行，可用：

ON ERR GOTO n      n是行号

使程序转向n行，使计算机继续运行。

## 二十六、逻辑判断

关系判断式也是逻辑判断式中的一种最简单的情况，关系判断式只有“成立”或“不成立”两种情况，我们把“成立”的情况叫做“真”，把“不成立”的情况叫做“假”。

“真”和“假”就是逻辑量。

逻辑判断式是由两个以上的逻辑量组成的，用逻辑运算符把它们联系起来，逻辑运算符有三个，即AND、OR、NOT。逻辑判断式的值，也只能是“真”或者“假”的两者之一。

1. AND 逻辑与。如 A AND B。

A和B都是“真”，同时成立，那么A AND B才是“真”、A和B是“与”的关系。

2. OR 逻辑或。如 A OR B

A或B只要有一个是“真”，二者有一个成立，那么A OR B就是“真”。A和B是“或”的关系。

3. NOT 逻辑非。如NOT A，A是“真”，NOT A就是“假”。NOT A是“非”的关系。

运用逻辑判断会给我们编写程序带来很大方便。常用的有AND、OR两个。

形式：IF 关系判断式或逻辑判断式 THEN语句1；  
语句2；语句3

例一，比较A，B，C值的大小，并打印出最大值。

```
10 INPUT A, B, C
20 IF A>B AND A>C THEN PRINT
   A:GOTO 100
30 IF B>A AND B>C THEN PRINT
   B: GOTO 100
40 PRINT C
100 END
```

例二，判断A，B，C三个值，A是否是最小值？

```
10 INPUT A, B, C
20 IF A>B OR A>C THEN PRINT A,
   "NOTMIN":GOTO 40
30 PRINT "MIN=", A
40 END
```

例三，某班期末考试四门课，其中两门为主课。请选出符合下列条件者，并打印出学号、总分和两门主课的成绩：

1.总分在300分或300以上；2.两门主课其中一门在90分以上。

设学号是W，主课一成绩是A，主课二成绩是B，另两门课成绩是C、D，总分是Y。

```
10 READ W, A, B, C, D
20 Y = A + B + C + D
```

```

30 IF W = 0 THEN END
40 IF Y >= 300 AND (A > 90 OR B > 90) THEN
    PRINT TAB (5) ; W ; TAB (10) ; Y ;
    TAB (15) ; A ; TAB (20) ; B ; GOTO 10
50 GOTO 10
60 DATA 3, 70, 60, 85, 90
70 DATA 8, 95, 80, 79, 98
80 DATA 10, 70, 90, 80, 65
  :
  :
500 DATA 0, 0, 0, 0, 0

```

说明：

1. THEN后可以写复合语句，语句间用符号：分开。
2. 如条件成立时，要把THEN后的所有语句执行完。如条件不成立，就继续执行这条语句的下一条语句。


## 二十七、循环的转出

循环可以有层嵌套，在APPLE II型机最多可嵌套10层循环，但在循环嵌套和转出中应注意以下几点：

在循环嵌套中除内外循环体不得交叉外，在循环语句和条件语句或转向语句联合使用时，可以从循环体内转到循环体外，但不允许从循环体外转入循环体内。多重循环只允许从内循环体转到外循环体，不允许从外循环体转入内循环体，多重循环只能经过FOR向下执行，不能从中间或从后向前执行。但是从循环体内转到循环体外执行一段，还可以再转回循环体内。

如：下图是正确的

下图是不正确的



## 二十八、APPLE I 型机彩色作图

APPLE I 型机的彩色作图分为低分辨率作图和高分辨率作图两种。

(一)、低分辨率作图语句：

1. GR

形式： GR

执行这条语句，计算机打开彩色通道，把屏幕置于低分辨率图形显示方式（ $40 \times 40$ ），屏幕下部留出4行用作正文显示。

如果需要变成满屏幕图形显示（ $40 \times 48$ ），就可输入 POKE-16302, 0或POKE-49324, 0。再要恢复通常方式可输入TEXT

2. COLOR = 数学表达式


低分辨率彩色作图共16种色彩，编号是0~15，数学表达式代表色彩号，色彩号如下：

0 黑色      1 品红      2 深蓝      3 紫

4 深绿	5 灰	6 天蓝	7 淡蓝
8 褐	6 橙	10 灰	11 粉红
12 绿	13 黄	14 湖蓝	15 白

如不送入这条语句，就按COLOR = 0对待。

2. PLOT 数学式 1, 数学式 2



数学式1是X坐标，数学式2是Y坐标，以此标出画点（色块）的位置。

X和Y值都要在0~39之间，画点的颜色由离PLOT最近的COLOR决定，同一画点上如给了两种颜色，后给的颜色要取代先给的颜色。

4. HLIN 数学式 1, 数学式 2 AT 数学式 3

数学式1和数学式2是X坐标，分别是X轴上的两点，数学式3是Y坐标。这个语句是要在Y行上，用色块画一水平线。

5. VLIN 数学式1, 数学式2 AT 数学式3

数学式1和数学式2是Y坐标，分别是Y轴上的两点，数学式3是X坐标。这个语句是要在X列上，用色块画一垂直线，例题：

```

10 GR
20 FOR I= 0 TO 15
30 COLOR=I
40 HLIN 0,39 AT I

```

```

50 VLIN 0,39 AT I
60 NEXT I
70 END

```

(二)、高分辨率作图

1. HGR HGR2  
 形式: HGR  
 HGR2


执行这条语句,使计算机打开彩色通道,把屏幕置于高分辨率图形显示方式。(HGR是 $280 \times 160$ ,下部留出四行用作正文显示。HGR2是 $280 \times 192$ ,是满屏幕)

2. HCOLOR = 数学表达式

高分辨率彩色作图共 8 种色彩,编号是 0~7,各种彩色显示器常有不同的颜色。

3. HPLOT 数学式1, 数学式2 TO 数学式3, 数学式4

即: n (行号) HPLOT X1, Y1 TO X2, Y2



X1、X2是X坐标, Y1、Y2是Y坐标,在这两点间画一线。垂直、水平、斜线都可。

X值要在 0 ~ 279 之间,

Y值要在 0 ~ 159 之间, 满屏幕显示要在 0 ~ 191 之间。

如果下一条线的起点是这条线终点时，在同一语句中可以写成：40 HPLOT X1,Y1 TO X2,Y2 TO X3,  
Y3 TO X4, Y4

如不同在一语句时，可以写成：

```
40 HPLOT X1, Y1 TO X2, Y2
```

```
50 HPLOT TO X3, Y3
```

```
60 HPLOT TO X4, Y4
```

例：

```
10 HGR
```

```
20 FOR I=0 TO 7
```

```
30 HCOLOR=I
```

```
40 HPLOT 10,10 TO 269,10 TO 269,149  
TO 10,149 TO 10,10
```

```
50 NEXT I
```

```
60 END
```

## 二十九、LASER310机作图

### (一) 黑白几何图形做图

LASER310型机可用打印字符串的办法，用几何图形作图。

### (二) 彩色作图：

LASER310型机彩色作图有两种模式，就是文字模式和彩色作图显示模式。有以下指令：

#### 1.MODE (0)

执行这个指令就是文件模式32字×16行的显示方式，也


可使用16种几何图形作图。如果要用点来显示图像，可有 $64 \times 32$ 点。

## 2. MODE (1)

执行这个指令就把屏幕置于作图显示模式，这种模式能显示 $128 \times 64$ 点。

在执行完用这种模式编出的程序后，计算机便会自动恢复到文字模式，即MODE (0)的模式，图像就会消失。如想继续保留图像，就需要用FOR.....NEXT或GOTO语句使这程序可为空循环。

如果在MODE (1)模式程序执行过程中需要文字显示模式，可在所需位置加入指令MODE (0)便可。



## 3. SET (X, Y)

X是横坐标，值要在 $0 \sim 127$ 之间。

Y是纵坐标，值要在 $0 \sim 63$ 之间。

这条指令要求我们把一点画在横坐标为X，纵坐标为Y的位置上。

如果我们需重画一条起点为0，终点为127，并在第45列

上的直线，则可编程序：

```
10 MODE(1)
20 COLOR 2
30 FOR I=1 TO 127
40 SET(I,45)
50 NEXT I
60 GOTO 60
70 END
```

#### 4. RESET (X, Y)

坐标X、Y值的范围同SET (X, Y) 中X、Y值的范围  
这条指令与SET指令正好相反，它是把SET指令画出的  
点清除掉，是去掉坐标(X, Y)位置上的点。

#### 5. COLOR I, J颜色

I是前境颜色，共8种，编号是1~8；J是背境颜色，共  
2种，编号是0、1。

在MODE(0)模式中，前境颜色彩号是：1绿、2  
黄、3蓝、4红、5浅黄、6浅蓝、7紫、8橙。

背境颜色彩号是：0绿、1橙。

如果将背境颜色变成橙色，需要键入COLOR, 1。如  
果要回到绿色，只要键入COLOR, 0就行了。

在MODE(1)模式中，背境颜色的色彩号是：0绿  
色、1浅黄。

如果背境颜色是绿色(0)，则前境颜色的色彩号是：  
1绿、2黄、3碍、4红。

如果背境颜色是浅黄(1)，那么前境颜色的色彩号  
是：5浅黄、6浅蓝、7紫、8橙。

例：10 COLOR 2, 0

它是使前境为黄色而背境为绿色。

10 COLOR 3, 1

它是使。境为蓝色，背境为浅黄。

10 COLOR 4

它是使背境为红色而背境保持不变。

10 COLOR, 0

它是使背境为绿色而前境则保持不变。

### 三十、LASER310机的音乐程序

形式：

SOUND数学表达式1，数学表达式2

数学表达式1是确定音高的代码，有效值范围是0~31。

除0表示休止符外，其余代码可与钢琴键盘相对应。



以C调为例  
 简谱 6 7 1 2 3 4 5 6 7 1 2 3 4 5 6 7 1 2  
 6 7 1 2 3 4 5 6 7 1 2 3 4 5 6 7 1 2

数学表达式2是确定时值的代码，有效值范围是0~9，如果我们以四分音符（以简谱“5”为例）做为一拍，代码是6，则其它音符代码是：

时值(拍) 5 5 5 · 5 5 · 5 5 · 5 - 5 - -  
 ≡ ≡

代码： 1 2 3 4 5 6 7 8 9

执行SOUND指令，计算机就按照数学表达式1指定的

频率和数学表达式 2 指定的时值，触动蜂鸣器发出声音。我们可以把乐曲编码连续地写成语句，计算机就会演奏这个乐曲。

例：乐曲《清晨》中的一段

1 = C2/4

$\begin{array}{cccc|cccc|} \overset{\cdot\cdot}{11} & \overset{\cdot\cdot}{21} & 7 & 5 & 6 & \overset{\cdot}{16} & \overset{\cdot}{36} & \overset{\cdot}{16} & | \\ \hline \end{array}$   
 $\begin{array}{cccc|cccc|} \overset{\cdot\cdot}{22} & \overset{\cdot\cdot}{21} & \overset{\cdot\cdot}{77} & \overset{\cdot\cdot}{12} & 5 & - & 5 & 0 & | \dots \\ \hline \end{array}$

程序是，

```

10 SOUND 16,4 编写第一个音符  $\overset{\cdot}{1}$ ，音高  $\overset{\cdot}{1}$ 
                        代码为16，二分之一拍代码
                        为4
20 SOUND 16,4 编写第二个音符  $\overset{\cdot}{1}$ 
30 SOUND 18,4 编写第三个音符  $\overset{\cdot}{2}$ 
40 SOUND 16,4 编写第四个音符  $\overset{\cdot}{1}$ 
50 SOUND 15,6 编写第五个音符 7
60 SOUND 11,6 编写第六个音符 5
70 SOUND      :      :
                        :      :

```

这种方法虽然易学，但编写和输入程序都很麻烦，我们可以改用循环、读数语句来编写这个程序

```

10 FOR I= 1 TO 22  这首共有22个音符
20 READ A, B      变量A读入音高代码

```

		变量B读入时值代码
30	SOUND A, B	按A, B的值发声
40	NEXT I	
50	DATA 16,4,16,4,18,4,16,4,15,6,	
	11,6,13,6,16,4,13,4,8,4,13,4	
60	DATA 16,4,13,4,18,4,18,4,18,4	
	16,4,17,4,15,4,16,4,18,4,11,9	
70	END	

如果想反复听这首歌，应首先恢复数据区，然后无条件转向循环语句：

```

45 RESTORE
47 GOTO 10

```

## 第三章 文件存取和打印机使用

### 一、文件存取

将主机内存中的程序，存到驱动器磁盘上备用，需用时再将磁盘上外存的程序调入主机使用。文件存取有以下指令：

#### (一) SAVE 文件名

执行这个指令可以把主机内存中的程序，以自定的文件名，存到接通的驱运器中的磁盘上，（磁盘需要先进行初始化，指令见后）。

文件名必须以英文字母开头，APPLE II 型机文件名最长不要超过30个符号，除符号外都可做文件名。LASER310机最长不要超过8个字符。文件名与SAVE当中要空一格，同一磁盘上不能以相同的文件名存入程序，否则后存入的将先存入的程序清除。

#### (二) LOAD 文件名

执行这个指令可以把接通的驱动器中磁盘上的以这一文件名存入的程序，调入主机内存。调入后，即将计算机内原内存程序清除。待驱动器红灯熄灭后，就可以送入LIST或RUN指令。

### (三) RUN 文件名

不再经LOAD文件名,就可将磁盘外存程序调入主机内存并执行一遍。

在LASER 310 型机磁盘驱动器上使用以上三条指令时,在指令后的文件名前后,要用“ ”引起来。

在LASER 310 型机磁带机上使用以上三条指令时需在指令前加一个字符,CSAVE、CLOAD、CRUN。

### (四) CATALOG

调入并显示磁盘外存程序目录。在CATALOG后可加D1或D2。D1、D2是驱动器编号。

目录中第一符号是\*。

第二个符号是文件属性。A表示浮点BASIC编写的文件, I表示整形BASIC文件, B表示二进制文件。

后面的数字表示文件所占磁盘区段数,表示所占容量的大小。

再后是文件名。

在有提示符>时,调用B文件要输入BRUN,BLOAD。如要恢复]符时,需输入FP。

LASER310型机调入磁盘外存程序目录的指令是DIR。如需要使用第二个驱动器,可输入DRIVE2;再使用第一个驱动器时,可输入DRIVE1。

### (五) DELETE 文件名

删去磁盘上这一文件名的文件程序。LASER310型机

用ERA“文件名”。

### (六) INIT

在初次存磁盘前必须使用这一指令将磁盘初始化。通常输入INIT后输入HELLO作为引导程序名。(APPLE I型机上使用INIT命令时必须加文件名作为自动引导程序)。

### (七) COPY

将一个磁盘上的全部文件复制到另一个磁盘上 不需再初始化。

LASER310型机复制文件的指令是DCOPY。计算机执行这一指令时，会在屏幕上显示出SOURCEDISK(1/2)?是问：原盘驱动器的号，是1号还是2号?可使用键盘输入原盘所在驱动器的号。接着屏幕上又要显示出DESTINATION DISK(2/1)?是问：复制盘驱动器的号，是2号还是1号?键盘输入后就开始复制。

在进行磁盘操作时，不得开关主机及与之相联系的所有外围设备的电源。

**二、打印机的使用。**本书只介绍APPLE I型机打印机的使用。

步骤是：

(一) 上纸。如在没有上纸或纸将用完时接通打印机，打印机就要发出声响，红灯亮，以示警告，这时不可打印，用户应给打印机上纸。

(二) 接通电源开关，指示灯亮。

(三) 当打印机卡插在1号槽时，输入PR# 1指令。

READY和ONLINE灯亮。

(四) 输入LIST指令, 可将主机内存程序结果在打印机上打印出来。

(五) 输入RUN指令。将程序运行结果打印出来。

(六) 输入PR# 0指令。使打印机与计算机主机脱机, 不能再打印主机内存程序。ONLINE灯灭。

(七) 撕纸或按打印机上LF键, 慢速走纸或转动纸轴走纸。尽量不要用FF键快速走纸。切不可向相反方向将纸拉出来。

(八) 当不用打印机时要关掉打印机的电源开关。

APPLE I 型机高分辨率作图图形需在图形显示停止不动时方可用打印机打印。打印时需输入PR # 1以后, 再同时按CTRL键和Q键, 才能打印出图形。

### 三、大小写

APPLE I 型机输入BASIC语言指令中不能用小写字母, 但如要在字符串中使用小写字母时, 需使用转换大小写的指令。有的计算机设有大小写转换键。有的计算机需同时按SHIFT CTRL和U键, 就可以输入小写字母。如需恢复输入大写字母时, 仍需再分别按以上这些键。LASER 310型机不能输入小写字母。

## 第四章 指法训练

计算机的键盘与英文打字机的键盘字母排列是相同的，它是根据英语语法的特点，每个单词拼写时字母出现的频率次数多少而顺序排列的，称作标准键盘，我们以APPLE II型计算机52个键（包括空格键）的键盘为例（见附图三），操作键盘要像弹钢琴一样，每个指头负责哪个键是有一定之规的。操作计算机的键盘，也和练习毛笔字一样要有良好的正确姿势，选择坐位与机台的高低要适当，以坐得舒适便於操作为度，坐姿端正，上身正直略前倾，全身放松，两腿与肩同宽，两脚平铺地面，稍稍用力保持全身平稳，两臂自然下垂，两手平伸，击键时手背与键盘平行，击键时力度要均匀，不可过於用力，击过键后手指立即收回，主要依靠手指关节的弯曲用力，尽量避免手腕用力，开始要认真、仔细、慢慢地练习，熟悉每个手指的分工，（见附图三键盘位置和手指和编写分工）切忌用单只手、一个手指去戳来戳去，这样精力不易集中，不仅不会提高速度，达不到熟练的程度，而且会顾此失彼搞得筋疲力尽。先掌握了每个手指所分工的各个字符，按照练习中的具体要求，再循序渐进、耐心刻苦地练习。初练时以每分钟40个字符的速度为宜，逐步提高速度。键盘操作熟练与否没有什么技术问题，只有靠刻苦地练习，是比较单调枯燥的，必须以滴水穿石、铁杵磨针的精神磨炼，只有具备了一定基础才能达到一个新的境界。俗话说“熟能生巧”，就是要苦练才能出真功夫。

为了便于操作，文稿的位置可按自己的习惯放置，一般可置于身体右侧，目视文稿，默念一段，然后转向目视屏幕，尽量不去看手指击键，要由视觉击键过渡到触觉击键，也即俗称的“盲打”，久而久之即可运用自如。

练习：

1. 按字母正向排列顺序ABCDEFGHIJKLMNO  
PQRSTUVWXYZ 反复练习、

2. 按字母反向排列顺序ZYXWVUTSRQPONM  
LKJIHGFEDCBA 反复练习、

3. 选择一篇英语短文，随读随击键，在屏幕上显示出来。

4. 选择一篇汉字短文读出汉字，用汉语拼音字母拼写，随读随击键，在屏幕上显示出来。

5. 选择总复习上的例题题解的任一程序用正确的指法击键输入计算机，最后运行得出结果。

要注意，每次练习要掌握循序渐进，计时检测记录下每分钟能达到的击键次数，除去屏幕显示的击键错误，一般能达到每分钟击键120个字符的速度就可以运用自如了。

## 第五章 复习提要

在本教材中，我们按照BASIC语言学习示意图 对基本BASIC语言和部分扩展BASIC语言横向进行了学习（汇编语言没有列入本教材当中），现在可以按以下提要进行复习。

### 一、基本BASIC有17条指令。

(一) 执行语句  
12条指令

输出语句：打印 PRINT

数据输入语句 { 赋值 LET  
键盘输入 INPUT  
读数 READ

转移 { 无条件转向 GO TO  
转子 GOSUB  
返回 RETURN

控制语句

条件判断 IF..... THEN  
循环 FOR  
...  
NEXT  
END  
停止、暂停 STOP

(二) 非执行语句  
5条指令

置数 DATA  
恢复数据  
区RESTORE

自定义函数 DEF FN

数组说明语句 DIM

注释 REM

## 二、扩展BASIC

执行语句	逻辑判断	IF THEN	GOTO
			GOSUB
	开关语句	ON I	GOTO
			GOSUB

## 三、键盘操作命令

NEW

HOME (APPLE I 型机)、CLS (LASER 310型机)

RUN

RUN n

LIST LISTn LIST n<sub>1</sub>,n<sub>2</sub> LIST -n LIST n<sup>m</sup>

CONT

ESC 键

CTRL 键

INT

FP

DEL n

BREAK 键 (LASER 301型机)

## 四、扩展键盘操作命令

SAVE CSAVE “ ” (LASER301型机)

LOAD CLOAD “ ” (LASER301型机)

CATALOG

DELETE  
INIT  
COPY  
PR # 6  
PR # 1  
PR # 0  
LIST    LLIST (LASER310机)  
RUN

CTRL键Q

五、参照BASIC语言学习示意图，按打印作图、数据输入、算术运算函数、转移、条件判断、循环、文件存取、打印机的调用、系统调用其他等十方面的功能，纵向复习各有关指令、符号。

附图一 APPLE I 型机键盘图 (D1033)

1	"	&	\$	%	8	,	(	)	∅	*	=	RES
1	2	3	4	5	6	7	8	9		:	-	ET

ESC	Q	W	E	R	T	Y	U	I	O	@	REPT	RETN
										P		URN

CTRL	A	S	D	F	G	H	J	K	L	+	←	→
										,		

SHIFT	Z	X	G	V	B	^	M	<	>	?	SHIFT
						N		,	.	/	

大小写  
转换

空格键

附图二 LASER310型机键盘图


C	S	A	V	E	C	L	O	A	D	CR	UN	VER	I	F	I	C	A	S	NEW	R	E	A	D	D	A	B	R	E	A	K	
1	!	2	"	3	#	4	\$	5	%	6	,	7	&	8	(	9	)	∅	@	—											

SIN ( COS ( TAN ( ATN ( LOG ( EXP ( SGN ( SQR ( ABS ( INT (

FOR	TO	NEXT	RETURN	THEN	ELSE	IF	INPUT	LET	PRINT	FUNCTION							
Q	<input type="checkbox"/> W	<input type="checkbox"/> E	<input type="checkbox"/> R	<input type="checkbox"/> T	<input type="checkbox"/> Y	<input type="checkbox"/> U	<input type="checkbox"/> I	<input type="checkbox"/>	[	P	]	R	E	T	U	R	N

MODE	(STEP	DIM	GOSUB	GOTO	CLS	REM	TAB	(	INSTR	RUBOUT	INVERSE					
CTRL	A	<input type="checkbox"/> S	<input type="checkbox"/> D	<input type="checkbox"/> F	<input type="checkbox"/> G	<input type="checkbox"/> H	<input type="checkbox"/> I	<input type="checkbox"/>	K	/	L	?	;	+	:	*

SHIFT	Z	<input type="checkbox"/> X	C	V	B	N	↑	M	←	→	↑	·	<	>	S	H	I	F	T
-------	---	----------------------------	---	---	---	---	---	---	---	---	---	---	---	---	---	---	---	---	---



## APPLE I 型机错误信息表

(一) 发生了一个错误以后, BASIC 返回到以 ] 提示符指定的命令且显示出闪烁的光标。程序文本和变量值均不破坏, 但程序不能够继续执行了。而且将所有的 GOSUB 和 FOR 循环计数置 0。

(二) 在程序运行过程中为了避免错误引起的中断。可以使用错误处理程序 ON ERR GOTO 语句。

(三) 在立即执行方式中发生错误时。不打印行标号。出错误信息的格式是:

立即执行模式                   ? × × ERROR

迟缓执行模式                   ? × × ERROR IN YY

在上面的两种情况里 “××” 是具体的错误标志  
“YY” 是迟缓执行模式发生错误的行标号。在这种模式下语句中的错误在没执行前不检查。

错误信息	说 明
CAN, CONTINUS	试图继续执行一个不存在的、或发生了错误后 的或从程序中删除了语句行后的程序
DIVISION BY ZERO	除数为 0 出错
ILLEGAL DIRECT	不能使用 INPUT、DEF FN、GEF 和 DATA 语句作为直接执行模式命令
ILLEGAL QUANTITY	对于算术函数或字符串函数传递的参数超出范 围, 在下面儿情况下可能发生这种类型的错 误: a. 负的数组下标(例如, LET A(-1, = 0 b. LOG 函数的自变量是负数或 0; c. SQR 函数的自变量是负值; d. AABA 是负值 B 不是整数; e. 在 MID \$LEFT \$RIGHT \$WAIT

	PEEK POKE TAB SPC ON...GOTO 或任何图形函数中使用了不合适的自变量
NEXT WITHOUT FOR	NEXT语句中的变量与仍然有效的FOR语句中的变量不对应, 或一个无名的NEXT对应任何一个仍然有效的FOR
OUT OF DATA	DATA中的数据不够读, 或者DATA中含有不合适的数据
OUT OF MEMORY	上面的任何一种情况都可能产生该类型的错误 程序太长数量太多FOR循环嵌套多于10层, GOSUB的嵌套多于24层; 形式很复杂的表达式; 括号嵌套于38层; 试图置LOMEM, 到太高; 试图置LOMEM低于, 现行值试图置HIMMEM, 太低
FORMULA TOO COMPLEX	为IF "x x" THEN执行的命令多于两个语句
REDIM, D ARRAY	在一个数组已经说明了之后, 遇到了对同一个数组的另一个说明语句, 这个错误经常发生的情况是: 已经给定了数组的非特指说明价10象 $A(I) = 3$ 其后又在程序中去说明该数组 $DIMA(100)$
RETURN WITHOUT GOSUB	执行RETURN语句没有相对应GOSUB语句
STRING TOO LONG	试图利用连续操作建立一个长于255个字符的字符串
BAD SUBSCRIPT	试图使用超出了数组说明的数组元素, 当使用和说明的数组维数不匹配时也会产生这个错误。 例, LET A(1, 1, 1) = Z但A已由DIM A(2, 2)说明
SYNTAX ERROR	语法错误, 表达式中的括号失配, 在语句行中的非法字符, 不正确的标点等
TYPE MISMATCH	值语句的左边是算术变量而右边是串, 或函数的自变量应是串却给出的是数字, 反之亦然
UNDEF' STATEMENT	试图GOTO GOSUB或THEN到不存在的语句标号去

UNDEF'D FUNCTION	试图使用没有定义的自定义函数
OVERFLOW	计算的结果大于BASIC提供的数值形式 如果发生向下溢出的情况，给出0作为结果并且不打印出错误信息，程序继续执行
WRITE PROTECTED	磁盘封口，书写保护

注：LASER 310型机错误信息与APPLE I型机基本相同。

# 习 题

(第一章)

一、求  $1 + 1 + 1 + 1 + \dots + 1 = ?$   
100个1

程序如下，请输入计算机

NEW ↵ (注：↵是按回车键符号)

10 LET S=0 ↵

20 FOR I=1 TO 100 STEP 1 ↵

30 LET S=S+1 ↵

40 NEXT I ↵

50 PRINT "S=" ; S ↵

60 END ↵

RUN ↵

二、一百块正方形水泥板，最小一块边长1米，每块依次增加一米，问这一百块水泥板的面积总和是多少？

即  $1 \times 1 + 2 \times 2 + 3 \times 3 + \dots + 100 \times 100 = ?$

程序如下，请输入计算机。

NEW ↵

10 LET S=0 ↵

20 FOR X=1 TO 100 ↵

30 LET S=S+X\*X ↵

40 NEXT X ↵

```
50 PRINT "S=" ; S↵
60 END↵
RUN↵
```

三、将下列程序输入，看显示出什么图形。

```
NEW↵
10 FOR I=1 TO 10↵
20 PRINT TAB (15-I) ; ↵
30 FOR J=1 TO 2*I-1↵
40 PRINT "*" ; ↵
50 NEXT J↵
60 PRINT↵
70 NEXT I↵
80 END↵
RUN↵
```

如将这个程序中第40句删去符号，或改为符号，看程序执行结果有什么变化？

四、用LASER310型机，将本书第75页的音乐程序输入，再输入RUN↵看有什么结果。

五、在教师辅导下，学习使用一些已编制好的应用程序，并练习使用键盘的指法。

## (第二章 一至十二节)

六、把下列各数字写成科学记数法：

1) 8 9 2 3 7 6 4

2) 0.0 0 0 0 0 0 0 0 0 0 3

七、把下列各数写成日常计数法：

1)  $8.76876E+08$

2)  $6.54321E-07$

八、把下列数学式用BASIC表达式表示：

1)  $\frac{80-16 \times 15}{16+57 \div 3}$       3)  $\frac{A}{B + \frac{C}{D + \frac{E}{X}}}$

2)  $A \cdot B \cdot \frac{e_1 e_2}{d^2}$

九、BASIC表达式  $(A+B) / (7 * D) - X * Y / 3 * W$  的数学式是什么？

十、设  $X=8$ ,  $Y=4.5$ 。如想在程序运行后打印出下面结果，程序应如何编写？请输入计算机检验。

$$X * Y = 8 * 4.5 = 36$$

十一、写出下列程序，输入计算机进行计算。

1)  $10^4 - 10^3$

2) 圆半径  $r=10$ ，求圆面积

十二、写出以下程序执行结果，输入计算机检验。

```
10 PRINT "A", "B", "C", "D", "E"
```

```
20 PRINT 2*2,3*3,4*4,5*5,6*6,-2,  
-3,-4
```

```
30 PRINT "X", "Y", "Z"
```

```
40 PRINT 0.2,0.3,0.4,-5,-6,-7,-8
```

```
50 END
```

十三、用字符串变量输入计算机打印汉语拼音“爱科学”

十四、改正或指出下列语句中的错误：

1) 10 LET A=7, B=8, C=9

2) 10 LET N+4=M+8

3) 10 LET P^2=4

十五、写出下列程序执行结果,输入计算机检验。请注意  
意变量值在程序执行过程中的变化。

1) 10 A=15

20 B=23

30 C=A

40 A=B

50 B=C

60 PRINT "A=" ; A

70 PRINT "B=" ; B

80 END

2) 10 A=15

20 B=23

30 C=A

40 D=B

50 A=D


60 B=C

70 PRINT "A=" ; A

80 PRINT "B=" ; B

90 END

十六、编程序输入计算机,打印以下图形:



十七、用标准打印格式打印五校（校名可自定）青少年计算机比赛优秀人数统计表，每校参加两试（可用汉语拼音或英文），要有表头，栏目包括“校名”、“一试”、“二试”等三栏，最后要有“合计”。

注：LASER310型机本题改为只参加一试，栏目中包“校名”、“优秀人数”。

（第二章 十三至十八节）

十八、下列语句有什么错误？

- 1) 10 FOR I=3 TO 5 STEP -1
- 2) 10 FOR I=1 TO 5 STEP 0.5
- ∴
- ∴

```

50 NEXT J
3) 10 FOR I=1 TO 4
    20 FOR J=1 TO 5
    30 PRINT J, I
    40 NEXT I
    50 NEXT J
    60 END
4) 10 TAB (10), "*"
    20 TAB (15), "*"
    30 TAB (20), "*"
    40 END

```

十九、写出下列程序执行结果并输入计算机检验。

```

1) 10 FOR X=4 TO 8 STEP 1
    20 PRINT X,
    30 NEXT X
    40 END
2) 10 FOR X=4.5 TO 8 STEP 2.5
    20 PRINT X
    30 NEXT X
    40 END
3) 10 FOR Y=-5 TO -15 STEP -1
    20 LET X=Y+1
    30 PRINT X,
    40 NEXT Y
    50 END
4) 10 X=2

```

```
20 FOR Y=5 TO 10 STEP 2
30 Z=Y*X
40 PRINT Z
50 NEXT Y
60 END
```

二十、下列程序循环几次？结果是什么？输入计算机检验。

```
1) 10 FOR J=15 TO 6 STEP 2
20 PRINT J
30 NEXT J
40 END


2) 10 FOR J=1 TO 5 STEP 1
20 PRINT J
30 NEXT J
40 PRINT J
50 END
```

二十一、编制程序用循环语句求1到100间奇数和，并输入计算机。

二十二、编制程序用循环语句求1到100间偶数和，并输入计算机。

二十三、打印出1到10之间各奇数的平方与各偶数的立方。

二十四、用循环语句打印出以下图形（第一个\*号在第10列）




二十五、打印九九乘法表

二十六、用循环语句编程序计算  $\frac{A}{A^2-1}$  的值

其中  $A=5\sim 15$  STEP 5

二十七、一批管子堆放如右图，共  $N$  层（ $N$  是自然数），编程序计算这堆管子共多少根。



二十八、自选一个打印表格题。

二十九、计算  $Y = \frac{X^2}{1!} + \frac{X^3}{2!} + \frac{X^4}{3!} + \frac{X^5}{4!} + \dots +$

$$\frac{X^{N+1}}{N!}$$

设 $X=2$ ，请打印出 $N$ 分别等于1、2、3、4、5、6、7、8、9、10时的 $Y$ 值

三十、小明想用“\*”号打印两条交叉的斜线，他编制了以下这个程序，但没有出现他预想的结果，请想一想这是什么原因？

```
10 FOR A=1 TO 21
20 PRINT TAB (5+A) ; "*" ; TAB
   (26-A) ; "*"
30 NEXT A
40 END
```

三十一、用GOTO语句编程打印  $(1 + \frac{1}{n})^n$

$(1 + \frac{1}{n})^{n+1}$ ,  $(1 - \frac{1}{n})^n$ ,  $(1 - \frac{1}{n})^{n-1}$  的值

可用INPUT语句输入任意十个正整数。

(第二章 第十九节至二十二节)

三十二、下列程序有什么错误？

```
1) 10 IF X ≥ 0 THEN 80
2) 10 IF X+Y ≠ 3*5 THEN 50
3) 10 INPUT X
   20 IF X < 0 THEN 100
   30 PRINT X
   40 END
```

三十三、编程序计算 $X=3.5$ ,  $X=-4.8$ 时 $Z$ 的值

$$X = \begin{cases} \frac{2X^2 + 3X + 1}{4X^2 + 4X + 2} & X > 0 \\ \frac{2X^2 + 3X + 1}{X^2 + 1} & X \leq 0 \end{cases}$$

三十四、编一程序，输入计算机，判断键盘输入的数是否大于100，打印出结果（编程序前要画框图）。

三十五、编一程序可以比较两数的大小（编程序前要画框图）。

三十六、试卷有26道题，答对一道题给8分，错一道题扣5分。某学生答完后总分是零。编制一程序计算这学生答对几道题，答错几道题。

三十七、输入十个数，打印出其中最大数。

三十八、输入五个数，把最大的数和它的顺序号打印出来。

三十九、输入十个数，统计出其中正数，负数和零的个数。

四十、下面算式中缺少一个十位数和一个个位数。试编一个程序，找出使算式成立的两个数，并输出正确的算式。

$$\boxed{\phantom{0000000000}} \times 3 + 2 \times \boxed{\phantom{00000000}} = 99$$

四十一、一个数列，它的头三个数是0，0，1，第四个数是前三个数之和。以后每个数都分别是它前三个数之和。请编一程序，打印出这个数列。直到第六十个数或数的值超过 $10^{16}$ 为止。

四十二、用X块正方体木块搭宝塔，要求第一层放一块，第二层放三块，第三层放五块……（每层比上层多两块）。

编一程序计算最多可搭几层？还剩几块？并用\*号代替木块打出示意图来（剩余木块不打印）。其中X是小于100的自然数。

四十三、有一个五位数，若在它前面写上一个7，得一六位数A，若在它后面写上一个7，得一六位数B，A是B的五倍。编制一个程序，计算出这五位数是多少。

四十四、用子程序编制程序打印出以下五个组各三个数，并列组中最大数。

- 1) 3, 19, 16
- 2) 12.6, 7.8, 9.1
- 3) -5, 1-, -7
- 4) 215, 309, 316
- 5) 0, -1, -2

四十五、下列程序运行结果是什么？

- 1) 10 READ A, B, C, D, E, F  
20 T = A + B + C + D + E  
30 U = T / F  
40 PRINT "U=" ; U  
50 DATA 8, 4  
60 DATA 6, 15, 4, 5  
70 END
- 2) 10 READ A, B  
20 DATA 1, 2, 3  
30 READ C, D, E  
40 RESTORE  
50 DATA 4, 5, 6, 7

```

60 READ F, G, H, I, J, K, L, M, N
70 RESTORE
80 DATA 8, 9, 10
90 PRINT A, B, C, D, E, F, G, H,
    I, J, K, L, M, N
100 END

```

四十六、编制一个把华氏温度变成摄氏温度的程序。F是华氏温度值，C是摄氏温度值。公式是：

$$C = \frac{5}{9} (F - 32)$$

求当F = 32, 60, 48, 82, 98.6时C值

四十七、下列程序执行结果是什么？请输入计算机检验。

```

10 FOR I=1 TO 5
20 READ A
30 PRINT TAB (I) ; A
40 NEXT I
50 RESTORE
60 PRINT TAB (I) ; A
70 DATA 1, 2, 1, 2, 1
80 END

```

如果把变量A改成A (I) ，结果是什么？

(第二章 第二十三节至二十四节)

四十八、三个角分别是30°、60°、45°，请打印出第一

个角的正弦、第二个角的余弦、第三个角的正切函数值。

四十九、宇宙飞船脱离地球引力影响（第二宇宙速度），速度必须等于或大于 $V = \sqrt{2gR}$ ，问V最小要达到多少米/秒？（ $R = 6371$ 公里  $g = 9.8$ 米/秒<sup>2</sup>）

五十、设 $y = |(\cos X + \operatorname{tg} X) \cdot (\ln 5 + \sqrt{X^2 + 2})|$ ，求y值（键盘输入X值）。

五十一、指出以下各式中的X值的范围：

- 1)  $\operatorname{INT}(X + 5) = 17$
- 2)  $\operatorname{INT}(100 * X + 0.5) / 100 = 8.91$
- 3)  $\operatorname{INT}(10 * X + 2) / 10 = 7.2$

五十二、写出以下范围随机数的BASIC表达式：

- 1) 0 ~ 10之间的。
- 2) 0 ~ 100之间的。
- 3) 30 ~ 40之间的。
- 4) 53 ~ 64之间，不包括53的随机整数

五十三、这个程序的执行结果是什么？

```
10 A=INT(RND(1)*12)+50
20 B=INT(RND(1)*20+62)
30 FOR I=1 TO 300
40 IF A=B THEN 80
50 NEXT I
60 PRINT "$$$ $"
70 GOTC 90
80 PRINT "####"
90 END
```

五十四、给计算机输入十个数，打印出其中奇数、偶数

及零的个数。

五十五、编制程序求两个正整数的最小公倍数。

五十六、编制程序打印出3~100间的所有的素数（只能被1和自己本身整除的数是素数）。

五十七、有一人上楼，如果一步迈2阶，余1阶；一步迈3阶，则余2阶；一步迈4阶，则余3阶；一步迈5阶，则正好走完，编一程序计算出该楼梯最少有多少阶？

五十八、为小学生编写十道小于100的正整数，并且数不能重复的加减法算术题。

五十九、编制程序求：

$y = \lg 18 + \lg 24 - \lg 26 + \lg \sin 15^\circ - \lg \cos 20$ （对数是以10为底的）。

六十、编程序打印

$$A = \begin{vmatrix} 4 & 5 & 6 \\ 7 & 8 & 9 \end{vmatrix}$$

六十一、编制程序能把若干个（如15、20、30……）任意给的数，按大小顺序排列打印出来。

六十二、有程序如下请写出运行结果，

```
10 DIM A (2, 2)
20 FOR I= 0 TO 2
30 FOR J= 0 TO 2
40 READ A (I, J)
50 C (I, J) = A (A (I, J) , A (J, I) )
60 PRINT C (I, J) ,
70 NEXT J
80 PRINT
```

```

90 NEXT I
100 DATA 1, 0, 2, 2, 2, 0, 0, 2, 0
110 END

```

六十三、全部输入以下程序、请注明需要键盘输入什么数和字符才能分别打印出下面的三个图形？

```

10 PRINT "N=" ;
20 INPUT N
30 PRINT
40 LET Y=37
50 IF N<=4 THEN 10
60 FOR I=4 TO N STEP 0.6
70 LET Y=Y+2.1
80 NEXT I
90 IF IN(TY+0.001)< > 58 THEN 10
100 PRINT "N $=" ;
110 INPUT N $
120 PRINT
130 IF N $ = "SIWQEG" THEN 170
140 IF N $ = "SIWQGE" TEHN 200
150 IF N $ = "SIWEGQ" THEN 230
160 GOTO 10
170 LET G=I-5
180 GOSUB 260
190 GOTO 250
200 LET G=I-6

```

```

210 GOSUB 260
220 GOTO 250
230 LET G=I-7
240 GOSUB 260
250 END
260 REM LOOP
270 FOR M=1 TO G
280 PRINT TAB(10);
290 FOR N=1 TO 5
300 PRINT "*";
310 NEXT N
320 PRINT
330 PRINT
340 NEXT M
350 RETURN

```

```

* * * * * * * * * * * * * * * *
* * * * * * * * * * * * * * * *
* * * * * * * * * * * * * * * *
          * * * * * * * * * * *
                    * * * * * *

```

六十四、一个长方体的长、宽、高分别是A、B、C。编制一个程序求至少要多少个这样的长方体才能组成一个正方体。

(第二章 第二十五节至二十七节)

六十五、某商店有十种商品。每种商品又有3~4个品

种，各售出数量如下，求各种商品售出数和总售出量。

种类/品种	1	2	3	4	合计
一	28	32	64	79	?
二	21	9	9	5	?
三	3	7	9	18	?
四	78	96	105	26	?
五	6	0	0	62	?
六	15	76	94	9	?
七	21	76	32	64	?
八	7	0	9	5	?
九	3	7	9	4	?
十	128	96	42	0	?
合计					?

六十六、有八根火柴，由计算机和小朋友轮流拿1根、2根，最多不能超过3根，也不许不拿，谁拿到最后一根谁就是胜利者。请小朋友先拿。用开关语句编一个程序，使计算机始终胜利。

六十七、阅读以下程序，写出运行结果：

```
10 FOR I=0 TO 4: X=0
20 J=I+1: Y=1
30 PRINT J, : X=X+1
40 IF J=5 THEN Y=-Y
50 J=J+Y
60 IF X=5 THEN 80
70 GOTO 30
80 PRINT: NEXT I
```

(第二章 第二十八节至三十节 第三章)

六十八、分别用APPLE II型机低分辨率、高分辨率、LASER310型机作图或写美术字。

六十九、用LASER310型机编制一音乐程序。

七十、练习文件存取。

七十一、练习打印机的使用。

习题参考答案

六、1)  $8,923764E+06$  2)  $3E-10$

七、1) 876876000 2) 0,00000654321

八、1)  $(80-16*15)/(16+57/3)$

2)  $A*B*E_1*E_2/D^2$

3)  $A/(B+C/(D+E/X))$

九、
$$\frac{A+B}{7D} - \frac{XYW}{3}$$

十、10 X = 8

20 Y = 4.5

30 PRINT "X\*Y=" , X, "\*", Y,  
"=" , X\*Y

40 END

十一、1)10 LET A=10^4 2)10 LET R=10

20 LET B=10^3 20 S=3.14159\*R^2

30 PRINT A-B 30 PRINT "S=";S

40 END 40 END

RUN  
9000

十二、	A	B	C
	D	E	
	4	9	16
	26	36	- 2
	- 3	- 4	
	X	Y	Z
	0.2	0.3	0.4
	- 5	- 6	- 7
	- 8		

RUN  
S = 314.159

LASER	A	B
310型机	C	D
	E	
	4	9
	16	25
	36	- 2
	- 3	- 4
	X	Y
	Z	
	0.2	0.3
	0.4	- 5
	- 6	- 7
	- 8	

十三、  
10 A \$ = "AI KE XUE"  
20 PRINT A \$  
30 END

十四、 1) 10 A=7:B=8:C=9  
2)、3) 等号左边不能是表达式

十五、 1) A=23  
B=15  
2) A=23  
B=15

```

十六、 10 PRINT "*****"
      20 PRINT "          *"
      30 PRINT "          *"
      40 PRINT "          *"
      50 PRINT "          *"
      60 PRINT "          *"
      70 PRINT "          *"
      80 PRINT "          *"
      90 PRINT "          *"
     100 PRINT "          *"
     110 PRINT "          *"
     120 PRINT "          *"
     130 PRINT "          *"
     140 PRINT "          *"
     150 PRINT "          *"
     160 PRINT "          *"
     170 PRINT "          *"
     180 PRINT "          *"
     190 PRINT "          *"
     200 PRINT "          *"
     210 PRINT "          *"
     220 END "          *"

```

```

十七、 10 PRINT "*JI SUAN JI JINGSAI
      20 PRINT YOU XIU TONG JI*"
      30 PRINT "....."

```

40 PRINT  
50 PRINT "XIAOMING", "ISHI", "2SHI"  
60 PRINT  
70 PRINT "....."  
80 A \$= "2ZHONG"  
90 B \$= "8ZHONG"  
100 C \$= "3XIAO"  
110 D \$= "ZHONG XIN XIAO XUE"  
120 E \$= "AI KE XIAO XUE"  
130 A1= 20  
140 B1= 19  
150 C1= 15  
160 D1= 15  
170 E1= 16  
180 A2= 11  
190 B2= 9  
200 C2= 10  
210 D2= 9  
220 E2= 11  
230 PRINT  
240 PRINT A \$, A1, A2  
250 PRINT  
260 PRINT B \$, B1, B2  
270 PRINT  
280 PRINT C \$, C1, C2  
290 PRINT

```

300 PRINT D$, D1, D2
310 PRINT
320 PRINT E$, E1, E2
330 PRINT
340 PRINT "....."
350 X1 = A1 + B1 + C1 + D1 + E1
360 X2 = A2 + B2 + C2 + D2 + E2
370 PRINT
380 PRINT "HE JI", X1, X2
390 END

```

注：LASER 310机：50语句改为PRINT  
 "XIAO MING", "YOU XIU"

180~220语句取消，240~320语句中A2、B2、C2、  
 D2、E2及前面一个符号，取消，360语句取消。380语句中  
 X2及前一个符号，取消。

- 十八、1) 初值、终值、初长之间关系不正确。  
 2) FOR后变量与NEXT后变量不一致。  
 3) 双重循环交叉。  
 4) 行号后TAB前缺PRINT.符号,应为符号;

- 十九、1) 4      5      6      2) 4.5  
       7      8                    7  
 3) - 4   - 5   - 6      4) 10  
       - 7   - 8   - 9            14  
       -10   -11   -12           18  
       -13   -14

- 二十、1) 一次。15                    2) 五次。1

2  
3  
4  
5  
6

二十一 10 X = 0  
20 FOR I = 1 TO 100 STEP 2  
30 X = X + 1  
40 NEXT I  
50 PRINT "X = ", X  
60 END

二十二 10 X = 0  
20 FOR I = 1 TO 100 STEP 2  
30 X = X + I + 1  
40 NEXT I  
50 PRINT "X = ", X  
60 END

二十三、10 FOR X = 1 TO 10 STEP 2  
20 Y = X + 1  
30 PRINT X ^ 2  
40 PRINT Y ^ 3  
50 NEXT X  
60 END

二十四、1)  
10 FOR I = 1 TO 4  
20 FOR J = 1 TO 5

```

30 PRINT TAB (10+I-1) ; "*" ;
40 NEXT J
50 PRINT
60 NEXT I
70 END
2) 10 FOR I= 1 TO 4
20 FOR J= 1 TO 5
30 PRINT TAB (10-I+1) ; "*" ;
40 NEXT J
50 PRINT
60 NEXT I
70 END
3) 10 FOR I= 1 TO 4
- 20 PRINT TAB (10-I+1) ; "*" ;
TAB (10+I) ; "*"
30 NEXT I
40 END
4) 10 PRINT TAB (10) ; "*"
20 FOR I= 1 TO 3
30 PRINT TAB (10-I) ; "*" ;
40 PRINT TAB (10+I) ; "*"
50 NEXT I
60 END
5) 10 PRINT TAB (10) ; "*"
20 FOR I= 1 TO 3
30 PRINT TAB (10-I) ; "*" ;

```

```

40 PRINT TAB (10+I) ; "*"
50 NEXT I
60 FOR I= 3 TO 2 STEP -1
70 PRINT TAB (10-I+1) ; "*" ;
80 PRINT TAB (10+I-1) ; "*" ;
90 NEXT I
100 PRINT TAB (10) ; "*"
110 END

```

6) 答案 (一)

```

10 FOR I= 1 TO 3
20 FOR J= 1 TO I* 2 - 1
30 PRINT TAB (10-I+1) ; "*" ;
40 NEXT J
50 PRINT
60 NEXT I
70 FOR J= 2 TO 1 STEP -1
80 FOR J= 1 TO 2*I-1
90 PRINT TAB (10-I+1) ; "*" ;
100 NEXT J
110 PRINT
120 NEXT I
130 END

```

答案 (二)

```

10 FOR I= 0 TO 2
20 FOR J= -I TO I
30 PRINT TAB (10-I) ; "*" ;

```

```

40 NEXT J
50 PRINT
60 NEXT I
70 FOR I=1 TO 0 STEP -1
80 FOR J=-I TO I
90 PRINT TAB (10-I) , "*" ;
100 NEXT J
110 PRINT
120 NEXT I
130 END

```

二十五、答案 (一)

```

10 FOR I=1 TO 9
20 FOR J=1 TO 9
30 PRINT I,"×",J,"=",I*J, " "
40 NEXT J
50 PRINT
60 NEXT I
70 END

```

答案 (二)

```

10 FOR I=1 TO 9
20 FOR J=1 TO I
30 PRINT I, "×", I, "=",I*J," ";
40 NEXT J
50 PRINT
60 NEXT I
70 END

```

二十六、

```
10 FOR A= 5 TO 15 STEP 5
20 PRINT "R=" ; A/(A^2-1)
30 NEXT A
40 END
```

二十七、

```
10 INPUT "N=" ; N
20 FOR I= 1 TO N
30 S=S+I+1
40 NEXT I
50 PRINT "S=" ; S
60 END
```

二十九、

答案 1

答案 2

10 Y= 0	10 Y= 0 : X= 2
20 X= 2	20 A= X : B= 1
30 B= 1	30 FOR N= 1
40 FOR N= 1 TO 10	TO 10
50 A= X * X ^ N	40 A= A * X
60 B= B * N	50 B= B * N
70 S= A / B	60 Y= Y + A / B
80 Y= Y + S	70 PRINT
80 PRINT "N=" ;	"N=" ; N,
N, "Y=" ; Y	"Y=" ; Y
100 NEXT N	80 NEXT N
110 END	90 END

三十、因在执行过程中， $(5 + A)$  的值会逐渐增加，

大于 (26-A)。


```
三十一、10 INPUT N
20 T=1/N
30 E= (1 + T) ^ N
40 F=E* (1 + T)
50 S= (1 - T) ^ (N - 1)
60 R=S* (1 - T)
70 PRINT N
80 PRINT E, F, R, S
90 PRINT
100 GOTO 10
110 END
```

三十二 1)  $\geq$  错误, 应为  $> =$  2)  $\neq$  错误, 应为  $<>$   
3) 20语句中 THEN 100, 但输个程序最大

行号是40, 没有100。

```
三十三、10 INPUT X
20 IF X>0 THEN 60
30 Z=(2*X^2+3*X+1)/(X^2+1)
40 PRINT "X=" , X, "Z=" , Z
50 GOTO 10
60 Z=(2*X^2+3*X+1)/4*X^2
+4*X+2)
70 PRINT "X=" , X, "Z=" , Z
80 GOTO 10
90 END
```

三十四、



```

10 INPUT X
20 IF X>100 THEN 50
30 PRINT "X<=100 X=" , X
40 GOTO 60
50 PRINT "X>100 X=" , X
60 GOTO 10
70 END
  
```

三十五、


```

10 PRINT "2SHU BI DA XIAO"
20 INPUT A, B
30 PRINT
40 IF A<B THEN 70
50 IF A>B THEN 90
60 GOTO 110
70 PRINT A, "<" , B
  
```

```

80 GOTO 120
90 PRINT A, ">", B
100 GOTO 120
110 PRINT A, "=", B
120 GOTO 10
130 END

```



三十六、

10 FOR I=1 TO 26

```

20 B = 26 - I
30 IF 8 * I - 5 * B <> 0 THEN 50
40 PRINT I, B
50 NEXT I
60 END

```

I是答对题数，B是错题数。

三十七、

```

10 INPUT A           送入第一数
20 FOR I = 2 TO 10   循环九次
30 INPUT B           送入第二数，共九个数
40 IF A > B THEN 60
50 A = B             大数存在A所在单元里
60 NEXT I
70 PRINT "ZUI DA    打印最大数
   SHU = ", A
80 END

```

三十八、

```

10 R = 1             第一个顺号赋给R
20 INPUT X           输入第一个数
30 FOR I = 2 TO 5    循环4次
40 INPUT Y           输入第二个数
50 IF Y < X THEN 80  输入四个数
60 X = Y             大数存在X所在
                     单元里
70 R = I             大数顺序号存在R
80 NEXT I            所在单元里

```

```

90 PRINT "X=" ; X      打印最大数
100 PRINT "R=" ; R     打印最大数顺序
                        号

```

```

110 END

```

### 三十九、

```

10 Z = 0                正数累计个数先置 0
20 F = 0                负数累计个数先置 0
30 L = 0                零数累计个数先置 0
40 FOR I = 1 TO 10
50 INPUT X
60 IF X <= 0 THEN 90
70 Z = Z + 1
80 GOTO 130
90 IF X = 0 THEN 120
100 F = F + 1
110 GOTO 130
120 L = L + 1
130 NEXT I
140 PRINT "Z=" ; Z, "F=" ; F,
      "L=" ; L
150 END

```

### 四十、

```

10 FOR I = 1 TO 9
20 FOR J = 0 TO 9
30 IF (I * 10 + 3) + (20 + J) <> 99
   THEN 50

```

```
40 PRINT I, "3+2", J, "= 99"  
50 NEXT J  
60 NEXT I  
70 END
```

四十一、

```
10 INPUT A, B, C  
20 FOR I=1 TO 20  
30 PRINT A, B, C  
40 A = A + B + C  
50 IF A > 1E+15 THEN 110  
60 B = B + C + A  
70 IF B > 1E+15 THEN PRINT  
    A: GOTO 110  
80 C = C + A + B  
90 IF C > 1E+15 THEN PRINT A,  
    B: GOTO 110  
100 NEXT I  
110 END
```

四十二、

```
10 INPUT X  
20 FOR I=1 TO 100  
30 X = X - (2 * I - 1)  
40 IF X < 0 THEN 70  
50 Y = X  
60 K = I  
70 NEXT I
```

```
80 PRINT K, "HANG", "SHENGYU", Y
85 PRINT
90 FOR I=1 TO K
100 PRINT TAB (20-I),
110 FOR J=1 TO 2*I-1
120 PRINT "*",
130 NEXT J
140 PRINT
150 NEXT I
160 END
```

四十三、

```
10 FOR I=10000 TO 99999
20 A=700000+I
30 B=I*10+7
40 IF A/B=5 THEN 60
50 GOTO 70
60 PRINT I
70 NEXT I
80 END
```

四十四、

```
10 INPUT "A=" , A
20 INPUT "B=" , B
30 INPUT "C=" , C
40 GOSUB 500
50 GOTO 10
60 END
```

```

500 IF A>B THEN 520
510 A = B
520 IF A>C THEN 550
530 PRINT C
540 RETURN
550 PRINT A
560 RETURN

```

四十五、

1.  $U = 7.4$

2. 1 2 3

4 5 1

2 3 4

5 6 7

8 9

四十六、

```

10 PRINT "F" , "C"

```

```

20 READ F

```

```

30 C = (5/9) * (F - 32)

```

```

40 PRINT F, C

```

```

50 GOTO 20

```

```

60 DATA 32, 60, 48, 82, 98, 6

```

```

70 END

```

四十七、

A改成A(I)结果是:

1

1

2

2

1	1
2	2
1	1
1	0

四十八、

```

1) 10 PRINT SIN (30 * 3.14159/180)
    20 PRINT COS(60 * 3.14159/180)
    30 PRINT TAN(45 * 3.14159/180)
    RUN
    0.5
    0.5
    1

```

四十九、

```

10 R = 6371 * 1000
20 G = 9.8
30 V = SQR(2 * R * G)
40 PRINT "V=" ; V, "M/S"
50 END
RUN
11174.596 M/S

```

五十、

```

10 INPUT "X=" ; X
20 Y = ABS((COS(X) + TAN(X)) * (LOG(5)
    + SQR(X^2 + 2)))
30 PRINT "Y=" ; Y
40 END

```

五十一、

- 1)  $12 \leq X < 13$
- 2)  $8.905 \leq X < 8.915$
- 3)  $7 \leq X < 7.1$

五十二、

- 1)  $RND(1) * 10$
- 2)  $RND(5) * 100$
- 3)  $RND(3) * (40 - 30) + 30$
- 4)  $INT(RND(1) * (64 - 53)) + 54$

五十三、

\$ \$ \$ \$

五十四、

```
10 J=0           J是累计奇数的个数
20 E=0           E是累计偶数的个数
30 L=0           L是累计零数的个数
40 FOR I=1 TO 10
50 INPUT X
60 IF X=0 THEN 120 先判断是零吗?
70 IF INT(X/2)=X/2 THEN 100 再判断
                           是偶数吗?

80 J=J+1
90 GOTO 130
100 E=E+1
110 GOTO 130
120 L=L+1
130 NEXT I
```

```
140 PRINT "J=" ; J, "E=" ; E, "L=" ;  
L
```

```
150 END
```

五十五、

```
10 INPUT A, B
```

```
20 J=1
```

```
30 X=A*I
```

```
40 Y=INT(X/B)*B
```

```
50 IF X=Y THEN 80
```

```
60 I=I+1
```

```
70 GOTO 30
```

```
80 PRINT A, B, X
```

X是最小公倍数

```
90 END
```

五十六、

```
10 FOR X=3 TO 100
```

```
20 FOR I=2 TO X-1
```

定下除数范围

```
30 IF INT(X/I)=X/I THEN 60
```

```
40 NEXT I
```

```
50 PRINT X
```

```
60 NEXT X
```

```
70 END
```

五十七、

```
10 X=5
```

```
20 IF INT((X+1)/3)<>(X+1)/3 THEN 50
```

```
30 IF INT((X+1)/4)<>(X+1)/4 THEN 50
```

```
40 PRINT X:GOTO 70
```

```
50 X=X+10
60 GOTO 20
70 END
```

RUN

35

五十八、

```
10 S=0:F=0
20 FOR I=1 TO 10
30 A=INT(RND(1)*100)
40 B=INT(RND(1)*100)
60 IF INT(I/2)=I/2 THEN 110
70 PRINT I, ", " ; A, "+" ; B, "=" ;
80 INPUT C
90 IF A+B<>C THEN 170
100 GOTO 150
110 IF A<B THEN GOSUB 500
120 PRINT I, ", " ; A, "-" ; B, "=" ;
130 INPUT C
140 IF A-B<>C THEN 170
150 S=S+10
160 PRINT "10" ; "GOOD" :GOTO 190
170 F=F+(-10)
180 PRINT "-10" ; "WRONG"
190 NEXT I
210 PRINT "ZONG FEN=" ; S+F
220 END
```

500 T = A : A = B : B = T

510 RETURN

五十九、

10 DEF FNA(X) = X \* 3.14159/180

20 DEF FNL(X) = LOG(X)/LOG(10)

30 LET Y = FNL(18) + FNL(24) - FNL(26)  
+ FNL(SIN(FNA(15))) - FNL(COS  
(FNA(20)))

40 PRINT Y

50 END

六十、

10 DIM A(2, 3)

20 FOR I=1 TO 2

30 FOR J=1 TO 3

40 READ A(I, J)

50 PRINT A(I, J);

60 NEXT J

70 PRINT

80 NEXT I

90 DATA 4, 5, 6, 7, 8, 9

100 END

六十一、

10 INPUT N

20 DIM A(N)

30 FOR I=1 TO N

40 READ A(I)

```

50 NEXT I
60 FOR I=1 TO N-1
70 FOR X=I+1 TO N
80 IF A(I)>A(X) THEN 120
90 B= A(I)
100 A(I) = A(X)
110 A(X) = B
120 NEXT X
130 NEXT I
140 FOR I=1 TO N
150 PRINT A(I)
160 NEXT I
170 DATA N个数
180 END

```

六十二、

0	1	0
0	0	1
2	0	1

六十三、

N=10	N=10	N=10
N \$=SIWEGQ	N \$=SIWQGE	N \$=SIWQEG

六十四、

```

10 DIM R(3)
20 FOR I=1 TO 3
30 INPUT R(I)
40 IF R(1)>=R(I) THEN 80

```

```

50 C = R(I)
60 R(I) = R(1)
70 R(1) = C
80 NEXT I
90 D = 0
100 D = D + R(1)
110 IF D/R(2) <> INT(D/R(2))
    THEN 100
120 IF D/R(3) <> INT(D/R(3))
    THEN 100
130 PRINT D * D * D / (R(1) * R(2) * R(3))
140 END

```

六十五、

```

10 DIM A(10, 4)
20 FOR I=1 TO 10
30 G=0
35 FOR J=1 TO 4
40 READ A(I, J)
50 G=G+A(I, J)
60 NEXT J
70 PRINT G
80 W=W+G
90 NEXT I
100 PRINT TAB(15), W
110 DATA 28, 32, 64, 79, ..... (40个售出数)
120 END

```

六十六、

```
10 INPUT X1
15 IF X1<>INT(X1)THEN 580
20 IF X1>3 OR X1<1 THEN 580
30 ON X1 GOTO 100, 200, 300
100 J1=3
110 GOTO 310
200 J1=2
210 GOTO 310
300 J1=1
310 S1=8-J1-X1
320 PRINT "X1=" ; X1, "J1=" ; J1,
      "S1=" ; S1
330 INPUT X2
335 IF X2<>INT(X2) THEN 580
340 IF X2>3 OR X2<1 THEN 580
350 ON X2 GOTO 400, 450, 500
400 J2=3
410 GOTO 550
450 J2=2
460 GOTO 550
500 J2=1
550 S2=4-J2-X2
560 PRINT "X2=" ; X2, "J2=" ; J2, "S2=" ;
      S2
570 PRINT "JI SUAN JI YING LEI "
```

```
      :GOTO 590
580 PRINT "WEI FAN GUI ZE! ":GOTO
      10
590 END
```

六十七、

```
 1  2  3  4  5
 2  3  4  5  4
 3  4  5  4  3
 4  5  4  3  2
 5  4  3  2  1
```

### 总复习练习程序

- [1] 编写一个程序，使它能计算并打印出2到100之间（包括2和100）的所有偶数，并能打印出这些偶数之和。

程序：

```
10 S = 0 : X = 0
20 IF X > 100 THEN 70 (控制终止)
```

```

30 X = X + 2 (从2开始每次加2)
40 S = S + X (求偶数累加和)
50 PRINT X, (打印每个偶数)
60 GOTO 30
70 PRINT "S=" ; S (打印这些偶数之和)
80 END

```

[2] 编写计算  $S = \frac{1}{1 \times 2} + \frac{1}{2 \times 3} + \frac{1}{3 \times 4} + \dots + \frac{1}{n(n+1)}$  的程序。

程序:

分析: 输入N值后, 即知需计算N个分式的和。

```

10 INPUT N
20 P = 0 : S = 0
30 P = P + 1 (分母的第一个因数)
40 IF N < P THEN 80 (当P > N时终止计算, 转去打印结果)
50 B = 1 / (P * (P + 1)) (每项分式值的计算)
60 S = S + B
70 GOTO 30 (进行多次运算)
80 PRINT "S=" ; S
90 END


```

[3] 在铁路运输中, 货物不超过30公斤时, 按每公斤0.15元收运费, 在超过30公斤重以上时, 超过30公斤的部分按每公斤0.25元收运费。如果x公斤货物运费为y元, 编一程序计算y的值,

分析: (1) 找出计算公式:

$$y = \begin{cases} 0.15 \times x & (\text{当 } x \leq 30 \text{ 时}) \\ 0.15 \times 30 + 0.25 (x - 30) & (\text{当 } x > 30 \text{ 时}) \end{cases}$$

(2) 画出框图



程序:

```

10 INPUT "X=" , X
20 IF X>30 THEN 50
30 Y=0.15 * X
40 GOTO 60
50 Y=0.15 * 30 + 0.25 * (X-30)
60 PRINT "Y=" , Y
  
```

70 END

RUN

X = 28

[4] 用下面的DATA语句

800 DATA 0, 0, 0, 1; 0, 1, 1, 0, 1,  
999 (数值999是终止标志)

编一程序计算0和1的个数, 并打印出标有文字说明的这两个数。

程序:

800 DATA 0, 0, 0, 1, 0, 1, 1, 0, 1,  
999

810 Z = 0; W = 0 (置零)

830 READ X

840 IF X = 999 THEN 900 (控制终止)

850 IF X = 0 THEN 880 (850和880语句累计0的个数)

860 W = W + 1 (不是0则必是1, 累计1的个数)

870 GOTO 830

880 Z = Z + 1

890 GOTO 830

900 PRINT Z, "NUMBER OF ZEROES"  
(0的数目)

910 PRINT W, "NUMBER OF ONES"  
(1的数目)

920 END

RUN

5 NUMBER OF ZEROES

4 NUMBER OF ONES

[5] 编写一个读取下面数据的程序，使之能计算出偶数的平方和与奇数的立方和并且打印出结果

100 DATA 12, 23, 45, 36, 24, 57,

4, 5, 7, 2, 999其中999是终止标志。

程序:

10 S = 0 : P = 0 (置零)

20 READ X

30 IF X = 999 THEN 70 (控制终止)

40 IF X = INT (X/2) \* X THEN 60

(是偶数转向60语句)

50 B = X ^ 3 : P = P + B : GOTO 20 (不是偶数则是奇数，将此数立方后，累加)

60 A = X \* X : S = S + A : GOTO 20 (将此数平方后累加)

70 PRINT "S = " , S (打印偶数平方和)

80 PRINT "P = " , P (打印奇数立方和)

90 END

100 DATA 12, 23, 45, 36, 24, 57, 4,

5, 7, 2, 999

[6] 编写一个程序，使它产生1到100(包括1和100之间的随机整数，检查每一个数值，当随机数等于53时，程序应指出是第多少个随机数多于53，并打印出这个数。

程序：


```
10 K = 0
20 R = INT (RND) (1) * 100) + 1 (产生题
   意要求的随机整数)
30 K = K + 1 (计次数)
40 IF R = 53 THEN 60
50 GOTO 20
60 PRINT "THE NUMBER 53 APPEARED
   IN TRIAL" , K (数字53出现在第K个随机
   数)
70 END
```

[7] 由键盘输入A、B、C三个数，编一程序，它能把这三个数按其值的大小顺序打印出来。

程序：

```
10 INPUT "A, B, C=" , A, B, C
20 IF A > B THEN 40
30 T = A : A = B : B = T
40 IF B > C THEN PRINT A, B, C : GOTO
   70
50 IF A > C THEN PRINT A, C, B :
   GOTO 70
60 PRINT C, A, B
70 END
```

框图:



[8] 利用下面的DATA语句:

```
300 DATA 3, 21, 4, 9, 6, 18, 4, 1,  
14, 19, 17, 1000,
```

(数值1000是终止标志, 它指出数据到此结束)

编写一个程序, 从其中找出最大的数并打印出来程序:

```
10 READ L (读入第一个数)
```

```
20 READ X (与60语句呼应读取其余各数)
```

```
30 IF X=1000 THEN 70 (控制终止)
```

```

40 IF X<=L THEN 20 (多次比较大小总把最
   大值贮在变量L中)
50 L=X
60 GOTO 20
70 PRINT L, "IS LARGEST" (最大)
80 END
300 DATA 3, 21, 4, 9, 6, 18, 4,
      1, 14, 19, 17, 1000
RUN
21 IS LARGEST

```

〔9〕用上题中的DATA语句，编一程序找出其中最大和最小的数，并打印出标有文字说明的这两个数。

分析：先读入第一个数，把它看成目前既是最大又是最小的数然后和其后诸数比较，通过单元互换总使S单元存数最小，L单元存数最大，直至读取比较完全部数值。

程序：

```

300 DATA 3, 21, 4, 9, 6, 18, 4, 1,
      14, 19, 17, 1000
360 READ L
370 S=L
380 READ X
390 IF X=1000 THEN 470
400 IF X>L THEN 450 (400和450语句的
   作用是依序进行判断比较，总把较大的数放
   在L中)

```

```

410 IF X<S THEN430
420 GOTO 380
430 S = X (410和430语句的作用是依序进行判断比
      较，总把较小的数放在S中)
440 GOTO 380
450 L = X
460 GOTO 380
470 PRINT S, "IS SMALLESTVALUE"
      (最小值)
480 EPRINT L, "IS LARGESTVALUE"
      (最大值)
490 END
RUN

```

```

1 IS SMALLEST VALUE
21 IS LARGEST VALUE

```

[10] 编写一个计算  $2 + 4 + \dots + 24$  的程序，并要求打印出计算结果。

```

程序: 10 S = 0
      20 FOR I = 2 TO 24 STEP 2
      30 S = S + I
      40 PRINT S
      50 NEXT I
      60 PRINT "S = ", S
      70 END

```

[11]  $T = 20!$  (即  $T = 1 \times 2 \times 3 \times 4 \times \dots \times 20$ )

程序:

```

10 T = 1
20 FOR X = 1 TO 20
30 T = T * X
40 NEXT X
50 PRINT "T=" , T
60 END

```

[12] 求任意十个数的平方和。

程序：

```

10 S = 0 (置零) , (累加时应先给变量S置零)
20 FOR I = 1 TO 10
30 READ X (读入数值)
40 S = S + X * X (累加各项平方和)
50 NEXT T
60 PRINT "S=" , S (打印结果)
70 END
80 DATA 3 , 8 , 15 , 12 , -7 , 4 , 82 ,
      81 , 9 , -11
      (任意的十个数)

```

循环控制变量I，在此程序中只起计循环次数作用，不参与计算。

[13] 求任意十个数的相乘积。

程序：

```

10 T = 1 (累乘时应给变量T先置1)
20 FOR I = 1 TO 10
30 READ X
40 T = T * X

```

```

50 NEXT I
60 PRINT "T=", T
70 END
80 DATA 15, 32, 7, -8, 42, 71, 4, 8,
      -5, 3

```

(十个任意数)

循环变量I, 只起计数作用, 不参与计算。

[14] 编写程序打印出下面的图形

```

(1)  * * * * *
      * * * * *
        * * * *
          * * *
            *
            * * *
              * * * *
                * * * * *
                  * * * * *
                    * * * * *

```

程序:

```

10 H=17:B=17:K=1
20 FOR I=1 TO 9
25 IF I=5 THEN H=1:K=K-(H
   -B)/2
30 IF I>5 THEN B=H:H=2*H
   -I:K=K-(H-B)/2
40 PRINT TAB(11+K)
60 FOR J=1 TO H
70 PRINT "*",
80 NEXT J
90 IF I>5 THEN 110

```

```

100 B=H : H= (H + 1) / 2 : K = (B-
      H) / 2 + K
110 PRINT
120 IF I= 5 THEN H= 2 : K=K- 1
130 NEXT I

```

(2)

```

      * * *   * * *
     * * *   * * *
    * * *   * * *
   * * *   * * *
  * * *   * * *
 * * *   * * *
* * *   * * *
* * * * * * * * * * * * * * * *
* * * * * * * * * * * * * * * *
* * * * * * * * * * * * * * * *

```

程序：

```

10 FOR I= 1 TO 7
20 PRINT TAB (11-I) ; " * * * " ;
30 PRINT TAB (14+I) ; " * * * " ;
40 NEXT I
50 PRINT
70 FOR J= 1 TO 3
80 PRINT TAB (6) ;
90 FOR K= 1 TO 17
100 PRINT " * " ;
110 NEXT K
120 PRINT
130 NEXT J

```

140 END

(3)

```
      *
     * *
    * * *
   * * * *
  * * * * *
 * * * * *
* * * * *
 * * * * *
  * * * *
   * * *
    * *
     *
      *
```

要求用键盘输入一个N值控制每边打星号的个数。

程序:

```
5 INPUT N
10 PRINT TAB(20); "*"
20 FOR I=1 TO N-1
30 PRINT TAB(20-I); "*" ;
   TAB(20+I); "*"
40 NEXT I
50 FOR I=N-2 TO 1 STEP-1
60 PRINT TAB(20-I); "*" ;
   TAB(20+I); "*"
70 NEXT I
80 PRINT TAB(20); "*"
90 END
```

(4)

```
      * * * * *
     * * * * *
    * * * * *
   * * * * *
  * * * * *
 * * * * *
* * * * *
 * * * * *
  * * * * *
   * * * * *
    * * * * *
     * * * * *
      * * * * *
```

用键盘输入相邻两边的星号数，打印出不同形状的平行四边形。

N表示打印的行数，M表示一行打印的个数。

A控制平行四边形倾斜的方向， $A > 0$ 时平行四边形下边向右斜， $A < 0$ 时下边向左斜。

$2 < N < 20$ ,  $2 < M < 20$

程序：

```
10 INPUT N, M, A
20 FOR I=1 TO N
30 PRINT TAB (16+SGN (A) * I) ,
40 IF I>1AND I<N THEN 85
50 FOR J=1 TO M
60 PRINT " * " ,
70 NEXT J
75 PRINT
80 GOTO 90
85 PRINT TAB (16+SGN (A) * I) ,
    " * " , TAB (16+SGN (A) * I+M-1) ,
    " * "
90 NEXT I
100 END
RUN
```

? 5, 10, 4,

```
 * * * * *
 *
 *
 *
 * * * * *

```


RUN

? 8, 18, - 4



RUN

? 18, 6, 6



RUN

? 12, 4, - 2

```

          * * * *
         *   *   *
        *     *   *
       *       *   *
      *         *   *
     *           *   *
    *             *   *
   *               *   *
  *                 *   *
 *                   *   *
* * * *
      *
     * *
    * * *

```

(5)

由每边相等数量的星号组成的三角形，用键盘输入N  
值控制每边的星号的个数。  $1 < n < 20$

```

程序：10 INPUT N
      20 FOR I=1 TO N
      30 PRINT TAB(21-I)
      40 IF I>1 AND I<N THEN 100
      50 FOR J=1 TO I
      60 PRINT "*",
      70 NEXT J
      80 PRINT
      90 GOTO 110
     100 PRINT "*", TAB(20+I-1), "*"
     110 NEXT I
     120 END
      RUN

```

? 3

```
  *
 * *
* * *
```

RUN

? 5

```
      *
     * *
    * * *
   * * * *
  * * * * *
```

(6)

```
      0 0
     0 0 0 0
    0 0 0 0 0 0
   0 0 0 0 0 0 0 0
  0 0 0 0 0 0 0 0 0 0
 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0
```

程序:

```
10 FOR I=1 TO 8
20 PRINT TAB (20-I) ,
30 FOR J=1 TO 2*I
40 PRINT "0" ;
50 NEXT J
60 PRINT
70 NEXT I
80 END
```

(7)

(7)

```
      0
     1 1 1
    2 2 2 2 2
   3 3 3 3 3 3 3
  4 4 4 4 4 4 4 4 4
 5 5 5 5 5 5 5 5 5 5
```

分析：此图形与（1）题的共同点是每行比一行多两个字符，不同点是打印数字而不是打印\*号，只需将（1）题的程序稍加改动即可。

程序：

```
10 FOR I=0 TO 5 (共打印六行)
20 PRINT TAB (21-I) ; (每行的第一个
   字符的列数)
30 FOR J=1 TO 2*I+1 (每行字符递加
   2)
40 PRINT I; (打印的字符)
50 NEXT J
60 PRINT (换行)
70 NEXT I
80 END
```

(8)

```
      1
     2 2 2
    3 3 3 3 3
   4 4 4 4 4 4 4
  5 5 5 5 5 5 5 5 5
  6 6 6 6 6 6 6 6
  7 7 7 7 7
  8 8 8
  9
```

程序：

```

10 FOR I= 1 TO 5
20 PRINT TAB (21-I) ,
30 FOR J= 1 TO 2*I-1
40 PRINT I,
50 NEXT J
60 PRINT
70 NEXT I
80 FOR I= 4 TO 1 STEP -1
90 PRINT TAB (21-I) ,
100 FOR J= 1 TO 2*I-1
110 PRINT 10-I,
120 NEXT J
130 PRINT
140 NEXT I
150 END

```

(9)

```

          1
         2 1 2
        3 2 1 2 3
       4 3 2 1 2 3 4
      5 4 3 2 1 2 3 4 5
     6 5 4 3 2 1 2 3 4 5 6
    7 6 5 4 3 2 1 2 3 4 5 6 7
   8 7 6 5 4 3 2 1 2 3 4 5 6 7 8
  9 8 7 6 5 4 3 2 1 2 3 4 5 6 7 8 9

```

程序：

```

10 FOR I= 1 TO 9
20 X= I
30 FOR J= 1 TO 2*I-1

```

```

40 PRINT TAB (10- I) , X,
50 IF J>=I , THEN 70
60 X=X - 1
65 GOTO 80
70 X=X + 1
80 NEXT J
90 PRINT
100 NEXT I
110 END

```

(10)

```

1 0 0 0 0 0 0 0 0 0
0 1 0 0 0 0 0 0 0 0
0 0 1 0 0 0 0 0 0 0
0 0 0 1 0 0 0 0 0 0
0 0 0 0 1 0 0 0 0 0
0 0 0 0 0 1 0 0 0 0
0 0 0 0 0 0 1 0 0 0
0 0 0 0 0 0 0 1 0 0
0 0 0 0 0 0 0 0 1 0
0 0 0 0 0 0 0 0 0 1

```

从左上角到右下角对角线为1 其余位置是 0 的图形。

程序：

```

10 FOR I=1 TO 10
20 PRINT TAB (10) ,
30 FOR J= 1 TO 10
40 IF I=J THEN PRINT "1" , :
   GOTO 70
60 PRINT "0" ,

```

```

70 NEXT J
80 PRINT
90 NEXT I
100 END

```

若要同时打印出两条对角线，只需加一句。

```

50 IF J=11-I THEN PRINT 1; :
    GOTO 70即可。

```

若只打印两条对角线而其余位置不打印任何符号，则只需把60句改成

```

60 PRINT " ";
(11)

```

```

          0
        1 1 1
      1 1 1 1 1
    2 2 2 2 2 2 2
  2 2 2 2 2 2 2 2
3 3 3 3 3 3 3 3 3
4 4 4 4 4 4 4 4 4 4 4
5 5 5 5 5 5 5 5 5 5 5 5
6 6 6 6 6 6 6 6 6 6 6 6
7 7 7 7 7 7 7 7 7 7
8 8 8 8 8 8 8
9 9 9
9

```

## 方法I

程序:

```
10 FOR I= 1 TO 10
20 PRINT TAB (20-I) ,
30 FOR J= 1 TO 2 * I-1
40 PRINT INT (I/2) ,
50 NEXT J
60 PRINT
70 NEXT I
80 FOR I= 9 TO 1 STEP - 1
90 PRINT TAB (20-I) ,
100 FOR J= 1 TO 2 * I- 1
110 PRINT 10-INT ( (I+ 1) /2) ;
120 NEXT J
130 PRINT
140 NEXT I
150 END
```

## 方法II

程序:

```
10 FOR I= 1 TO 19
20 PRINT TAB (ABS (10-I) +10)
30 LET B= 2 * I- 1
40 IF I>10 THEN B= 2 * (19-I)
   + 1
50 FOR J= 1 TO B
60 PRINT INT (I/2) ;
```

```

70 NEXT J
80 PRINT
90 NEXT I
100 END

```

[15]求所有能被11整除，且不含重复数字的三位数，这样的三位数有多少？

程序：

```

10 FOR A= 1 TO 9
20 FOR B= 0 TO 9
30 FOR C= 0 TO 9
40 LET D=100 * A +10 * B +C
50 IF D/11 =INT (D/11) THEN
70
60 GOTO 100
70 IF A < > B AND A < > C
AND B < > C THEN 90
80 GOTO 100
90 PRINT D,
100 NEXT C
110 NEXT B
120 NEXT A
130 END

```

RUN

132	143	154
165	176	187
198	209	231

253	264	275
286	297	308
319	341	352
374	385	396
407	418	429
451	462	473
495	506	517
528	539	561
572	583	594
605	627	638
649	671	682
693	704	715
726	748	759
781	792	803
814	825	836
847	869	891
902	913	924
985	946	957
968		



[General Information]

书名=跟我学BASIC

作者=

页数=200

SS号=11541633

DX号=

出版日期=

出版社=

书名

前言

目录

第一章 电子计算机的概述

第二章 BASIC语言和程序设计

一、语言、程序

二、BASIC语言学习示意图

三、运算符号

四、数、变量、数学表达式

五、运算规则

六、NEW

七、HOME (CLS) 清屏

八、RUN执行、运行

九、LIST打印程序清单

十、PRINT打印语句

十一、LET赋值语句

十二、修改

十三、FORNEXT循环语句

十四、框图

十五、REM注释语句

十六、TAB (X) 打印格式语句

十七、INPUT键盘输入语句

十八、GOTO无条件转向语句

十九、IFTHEN条件判断语句

二十、子程序GOSUB转语句RETURN返回语句

二十一、READDATA读数置数语句

二十二、RESTORE恢复数据区语句

二十三、函数（三角函数、开平方根函数、随机函数、取整函数、绝对函数、符号函数、对数函数、指数函数、自定义函数）

二十四、DIM数组说明语句

二十五、ON开关语句

二十六、逻辑判断

二十七、循环的转出

二十八、APPLE 型机彩色作图

二十九、LASER310机作图

三十、LASER310机的音乐程序

第三章 文件存取和打印机使用

第四章 指法训练

第五章 复习提要

附图一 APPLE 型机键盘图

附图二 LASER310型机键盘图

APPLE 型机错误信息表

习题

习题参考答案

总复习练习程序

附图三 指法训练图示意